
Parallel Programming
DELPHI
with OmniThreadLibrary
Primož Gabrijelčič
thedelphigeek.org

Parallel Programming with OmniThreadLibrary

Primož Gabrijelčič

This book is for sale at http://leanpub.com/omnithreadlibrary

This version was published on 2015-08-19

Tweet This Book!

Please help Primož Gabrijelčič by spreading the word about this book on Twitter!

The suggested hashtag for this book is #omnithreadlibrary.

Find out what other people are saying about the book by clicking on this link to search for this hashtag on Twitter:

https://twitter.com/search?q=#omnithreadlibrary

Also By Primož Gabrijelčič

A Smart Book

Contents

Sample Book

This is just the sample of the real book. It includes selected parts from some chapters.

To check the status of the full book, visit the The Delphi Geek[1] blog.

You can buy the book on the LeanPub[2].

[1] http://thedelphigeek.org
[2] http://leanpub.com/omnithreadlibrary

Credits

The author would like to thank Gorazd Jernejc (also known as GJ) for his contribution to the OmniThreadLibrary project. Gorazd, OTL would not be the same without you!

Parts of code were provided by Anton Alisov and Lee Nover. Thanks!

Following programmers (in alphabetical order) helped with reporting and fixing bugs: ajasja, andi, Anton Alisov, dottor_jeckill, geskill, Hallvard Vassbotn, Jamie, M.J. Brandt, Mason Wheeler, Mayjest, meishier, morlic, Passella, Qmodem, Unspoken, Zarko.

Great thanks go to Pierre le Riche who wrote beautiful FastMM[3] memory manager and allowed me to include it in the distribution.

Cover page (c) Dave Gingrich, http://www.flickr.com/photos/ndanger/2744507570/

[3] http://sourceforge.net/projects/fastmm/

Introduction

This is a book about OmniThreadLibrary[4], a multithreading library for Embarcadero Delphi rapid development environment.

The book assumes that the reader has some understanding about the multithreading programming. If you are new to multithreading, an excellent introduction is *Multithreading - The Delphi Way*[5] by Martin Harvey. That book is oldie, but goldie.

A more up-to-date overview of Delphi multithreading capabilities is published in the *Delphi XE2 Foundations, Part 3*[6] by Chris Rolliston, which is available on Amazon.

Formatting Conventions

This books covers the latest official OmniThreadLibrary release.

When a part of the book covers some other version, a [version tag] in superscript will indicate relevant version or versions. Version numbers (f.i. `2.1`) are used for older releases and SVN revision numbers (f.i. `r1184`) are used for functionality that was added after the last official release.

A single version or revision number (f.i. [r1184]) indicates that the topic in question was introduced in this version and that it is still supported in the current release.

A range of two versions (f.i. [1.0-1.1]) indicates that the topic was introduced in the first version (`1.0` in this example) and that it was supported up to the second version (`1.1`). After that, support for this topic was removed or it was changed so much that an additional section was added to describe the new functionality.

Work in Progress

This book is a work in progress - it is being published as it is written. When you buy this book, you are not only buying it 'as is' - in accordance with the Lean Publishing Manifesto[7] you will get all subsequent ebook versions for free, forever.

To check the status of the book and to influence the writing process by voting on the importance of particular topics, visit the The Delphi Geek[8] blog.

---

[4] http://www.omnithreadlibrary.com
[5] http://thaddy.co.uk/threads/
[6] http://delphifoundations.com/
[7] http://leanpub.com/manifesto
[8] http://thedelphigeek.com

Learn More

A good way to learn more about the OmniThreadLibrary is to go through the demos. They are included in the standard OmniThreadLibrary distribution so you should have them on the disk already if you have installed OmniThreadLibrary. Each demo deals with very limited subset of OmniThreadLibrary functionality so they are farily easy to understand.

From time to time I'm presenting OmniThreadLibrary in live webinars. Recordings are available on Gumroad[9].

I am also frequently posting about the OmniThreadLibrary on my blog. All the relevant articles are tagged with the OmniThreadLibrary tag[10].

Support is also available on StackOverflow[11] (make sure to tag the question with `omnithreadlibrary`) and on the Google+ OmniThreadLibrary community[12].

You can also stay in touch by following my Twitter[13] account.

Advertisement

I teach various Delphi and Smart Mobile Studio topics. I'm also available for Delphi or Smart Mobile Studio consulting. Find more at http://primoz.gabrijelcic.org.

[9] https://gumroad.com/thedelphigeek
[10] http://www.thedelphigeek.com/search/label/OmniThreadLibrary
[11] http://www.stackoverflow.com
[12] https://plus.google.com/communities/112307748950248514961https://plus.google.com/communities/112307748950248514961
[13] http://twitter.com/thedelphigeek

Release Notes

2015-08-19

- Documented TOmniMessageQueue.
- Documented TOmniTwoWayChannel.
- Documented TOmniValueContainer.
- Documented TOmniCounter.
- Documented TGp4AlignedInt and TGp8AlignedInt.
- Documented TOmniRecordWrapper.
- Documented IOmniAutoDestroyObject.
- Documented IOmniEnvironment.
- Documented examples.

2015-07-29

- Documented TOmniValue.
- Documented TWaitFor.
- Documented TOmniLockManager.
- Adapted to the 3.04b release.
  - Documented TOmniSingleThreadUseChecker.
  - Adapted Communication Subsystem documentation.

2015-07-12

- Adapted to the 3.04 release.
  - Documented Parallel.For.
  - Documented Parallel.Map.
  - Documented TOmniBlockCollection.ToArray.
  - Added longer example for `Parallel.For`.
  - Documented IOmniTaskConfig.`SetPriority` and `.ThreadPool`.
  - Documented Parallel.Pipeline.`PipelineStage[]` property.
  - Documented new demos.
  - Documented other changes and additions.
- Updated hyperlinks.

2013-08-11

- Completed “Low-level Multithreading” chapter.

2012-12-22

- Completed “Synchronization” chapter.

2012-10-08

- Adapted to the 3.02 release.
  - Documented background worker initializers and finalizers.
  - Documented Async/Await.
- Fixed minor problems found by [jachguate].

2012-08-23

- Completed “How-to” chapter.
- Added “Release Notes”.
- Fixed images and linebreaks in the “High-level Multithreading” chapter.

2012-07-08

- Completed “Introduction to OmniThreadLibrary” chapter.

2012-07-05

- Completed “High-level Multithreading” chapter.

2012-06-12

- First release.

1. Introduction to OmniThreadLibrary

OmniThreadLibrary[1] is a multithreading library for Delphi, written mostly by the author of this book (see Credits for full list of contributors). OmniThreadLibrary can be roughly divided into three parts. Firstly, there are building blocks that can be used either with the OmniThreadLibrary threading helpers or with any other threading approach (f.i. with Delphi's TThread or with AsyncCalls[2]). Most of these building blocks are described in chapter Miscellaneous, while some parts are covered elsewhere in the book (Lock-free Collections, Blocking collection, Synchronization).

Secondly, OmniThreadLibrary brings low-level multithreading framework, which can be thought of as a scaffolding that wraps the TThread class. This framework simplifies passing messages to and from the background threads, starting background tasks, using thread pools and more.

Thirdly, OmniThreadLibrary introduces high-level multithreading concept. High-level framework contains multiple pre-packaged solutions (so-called *abstractions*; f.i. parallel for, pipeline, fork/join ...) which can be used in your code. The idea is that the user should just choose appropriate abstraction and write the worker code, while the OmniThreadLibrary provides the framework that implements the tricky multithreaded parts, takes care of synchronisation and so on.

1.1 Requirements

OmniThreadLibrary requires at least Delphi 2007 and doesn't work with FreePascal. The reason for this is that most parts of OmniThreadLibrary use language constructs that are not yet supported by the FreePascal compiler.

High-level multithreading framework requires at least Delphi 2009.

OmniThreadLibrary currently only targets Windows installation. Both 32-bit and 64-bit platform are supported.

1.2 License

OmniThreadLibrary is an open-sourced library with the OpenBSD license.

[1] http://www.omnithreadlibrary.com
[2] http://andy.jgknet.de/blog/?page_id=100

In short, this means that:

1. You can use the library in any project, free, open source or commercial, without having to mention my name or the name of the library anywhere in your project, documentation or on the web site.
2. You can change the source for your own use. You can also put a modified version on the web, but you must not remove my name or the license from the source code.
3. I'm not guilty if the software blows in your face. Remember, you got OmniThreadLibrary for free.

In case your company would like to get support contract for the OmniThreadLibrary, please contact me.

1.3 Installation

1. Download the last stable edition (download link is available at the OmniThreadLibrary site[3], or download the latest state from the repository[4]. Typically, it is safe to follow the repository trunk as only tested code is committed. [Saying that, I have to admit that from time to time a bug or two do creep in but they are promptly exterminated].
2. If you have downloaded the last stable edition, unpack it to a folder.
3. Add the folder where you [unpacked last stable edition/checked out the SVN trunk] to the Delphi's *Library path*. Also add the *src* subfolder to the *Library path*. In case you are already using units from my GpDelphiUnits[5] project, you can ignore the copies in the *src* folder and use *GpDelphiUnits* version.
4. Add necessary units to the `uses` statement and start using the library!

[3] http://www.omnithreadlibrary.com/omnithreadlibrary/download.htm
[4] https://github.com/gabr42/OmniThreadLibrary
[5] https://github.com/gabr42/GpDelphiUnits

If you have XE8 or newer, you can download and install OmniThreadLibrary with the built-in GetIt! package manager. In that case you don't have to manually add OmniThreadLibrary paths and you can also skip the next step (*Installing Design Package*). Both will be done automatically by the GetIt!

1.3.1 Installing Design Package

OmniThreadLibrary includes one design-time component (`TOmniEventMonitor`) which may be used to receive messages sent from the background tasks and to monitor thread creation/destruction. It is used in some of the demo applications.

To compile and install the package containing this component, follow these steps:

- From Delphi, open *packages* subfolder of the OmniThreadLibrary installation and select file *OmniThreadLibraryPackages{VER}.groupproj* (where *{VER}* indicates the version of the Delphi you're using; at the moment of writing *{VER}* could be *2007*, *2009*, *2010*, *XE*, *XE2*, *XE3*, *XE4*, *XE5*, *XE6*, *XE7*, or *XE8*).
- In the Project Manager window you'll find two projects – *OmniThreadLibraryRuntime{VER}.bpl* and *OmniThreadLibraryDesigntime{VER}.bpl*. If the Project Manager window is not visible, select *View*, *Project Manager* from the menu.



- Right-click on the *OmniThreadLibraryRuntime{VER}.bpl* and select *Build* from the pop-up menu.
- Right-click on the *OmniThreadLibraryDesigntime{VER}.bpl* and select *Build* from the pop-up menu.
- Right-click again on the *OmniThreadLibraryDesigntime{VER}.bpl* and select *Install* from the pop-up menu.
- Delphi will report that the `TOmniEventMonitor` component was installed.



- Close the project group with *File*, *Close All*. If Delphi asks you whether to save modified files, choose *No*.

You should repeat these steps whenever the OmniThreadLibrary installation is updated.

1.4 Why Use OmniThreadLibrary?

OmniThreadLibrary approaches the threading problem from a different perspective than TThread. While the Delphi's native approach is oriented towards creating and managing threads on a very low level, the main design guideline behind OmniThreadLibrary is: "Enable the programmer to work with threads in as fluent way as possible." The code should ideally relieve you from all burdens commonly associated with multithreading.

OmniThreadLibrary was designed to become a "VCL for multithreading" – a library that will make typical multithreading tasks really simple but still allow you to dig deeper and mess with the multithreading code at the operating system level. While still allowing this low-level tinkering, OmniThreadLibrary allows you to work on a higher level of abstraction most of the time.

There are two important points of distinction between TThread and OmniThreadLibrary, both explained further in this chapter. One is that OmniThreadLibrary focuses on tasks, not threads and another is that in OmniThreadLibrary messaging tries to replace locking whenever possible.

By moving most of the critical multithreaded code into reusable components (classes and high-level abstractions), OmniThreadLibrary allows you to write better multithreaded code faster.

1.5 Tasks vs. Threads

In OmniThreadLibrary you don't create *threads* but *tasks*. A task can be executed in a new thread or in an existing thread, taken from the thread pool.

A task is created using `CreateTask` function, which takes as a parameter a global procedure, a method, an instance of a `TOmniWorker` class (or, usually, a descendant of that class) or an anonymous method (in Delphi 2009 and newer). `CreateTask` returns an `IOmniTaskControl` interface, which can be used to control the task. A task is always created in suspended state and you have to call `Run` to activate it (or `Schedule` to run it in a thread pool).



The task has access to the `IOmniTask` interface and can use it to communicate with the owner (the part of the program that started the task). Both interfaces are explained in full detail in chapter Low-level multithreading.

The distinction between the task and the thread can be summarized in few simple words.

> *Task is part of code that has to be executed.*
>
> *Thread is the execution environment.*

You take care of the task, OmniThreadLibrary takes care of the thread.

1.6 Locking vs. Messaging

I believe that locking is evil. It leads to slow code and deadlocks and is one of the main reasons for almost-working multithreaded code (especially when you use shared data and forget to lock it up). Because of that, OmniThreadLibrary tries to move as much away from the shared data approach as possible. Cooperation between threads is rather achieved with messaging.

If we compare shared data approach with messaging, both have good and bad sides. On the good side, shared data approach is fast because it doesn't move data around and is less memory intensive as the data is kept only in one copy. On the bad side, locking must be used to access data which leads to bad scaling (slowdowns when many threads are accessing the data), deadlocks and livelocks.

The situation is almost reversed for the messaging. There's no shared data so no locking, which makes the program faster, more scalable and less prone to fall in the deadlocking trap. (Livelocking is still possible, though.) On the bad side, it uses more memory, requires copying data around (which may be a problem if shared data is large) and may lead to complicated and hard to understand algorithms.

OmniThreadLibrary uses custom lock-free structures to transfer data between the task and its owner (or directly between two tasks). The system is tuned for high data rates and can transfer more than million messages per second. However, in some situations shared data approach is necessary and that's why OmniThreadLibrary adds significant support for synchronisation.

> Some would disagree with the OmniThreadLibrary communication structures being called *lock-free*. In reality, there are no communication mechanisms that would correctly work in a multithreaded world without using locking. The name *lock-free* only implies that no *operating system* locking primitives are being used. Instead, OmniThreadLibrary achieves the thread-safeness by using *bus locking*, a special processor instruction prefix that achieves atomic operation in a multiprocessor system. Bus-locked operations are much slower than the normal assembler code, especially as they may stop other cores for the time of the operation execution, but then they are also much faster than the operating system locking.

Lock-free (or *microlocked*) structures in OmniThreadLibrary encompass:

- bounded (size-limited) stack
- bounded (size-limited) queue
- message queue
- dynamic (growing) queue
- blocking collection

OmniThreadLibrary automatically inserts two bounded queues between the task owner (`IOmniTaskControl`) and the task (`IOmniTask`) so that the messages can flow in both directions.



1.7 TOmniValue

A `TOmniValue` (part of the *OtlCommon* unit) is data type which is central to the whole OmniThreadLibrary. It is used in all parts of the code (for example in a communication subsystem) when type of the data that is to be stored/passed around is not known in advance.

It is implemeted as a smart record (a record with functions and operators) which functions similary to a `Variant` or `TValue` but is faster[6][7]. It can store following data types:

- simple values (byte, integer, char, double, ...)
- strings (Ansi, Unicode)
- Variant

[6] http://www.thedelphigeek.com/2010/03/speed-comparison-variant-tvalue-and.html
[7] http://www.cromis.net/blog/2010/03/tvalue-and-other-variable-like-implementation-tests/

- objects
- interfaces
- records (in D2009 and newer)

In all cases ownership of reference-counted data types (strings, interfaces) is managed correctly so no memory leaks can occur when such type is stored in a `TOmniValue` variable.

Two kinds of floating-point numbers can be stored in a `TOmniValue` – `double` and `extended`. Former are stored directly in the `TOmniValue` record while latter are wrapped in an `TOmniExtendedData` record, which increases memory usage and decreases performance. The use of `double` floating-point numbers is therefore recommended.

The `TOmniValue` type is too large to be shown in one piece so I'll show various parts of its interface throughout this chapter.

1.7.1 Data Access

The content of a `TOmniValue` record can be accessed in many ways, the simplest (and in most cases the most useful) being through the `AsXXX` properties.

```
1  property AsAnsiString: AnsiString;
2  property AsBoolean: boolean;
3  property AsCardinal: cardinal;
4  property AsDouble: Double;
5  property AsDateTime: TDateTime;
6  property AsException: Exception;
7  property AsExtended: Extended;
8  property AsInt64: int64 read;
9  property AsInteger: integer;
10 property AsInterface: IInterface;
11 property AsObject: TObject;
12 property AsOwnedObject: TObject;
13 property AsPointer: pointer;
14 property AsString: string;
15 property AsVariant: Variant;
16 property AsWideString: WideString;
```

Exceptions can be stored through the `AsObject` property, but there's also a special support for `Exception` data type with its own data access property `AsException`. It is extensively used in the *Pipeline* high-level abstraction.

While the setters for those properties are pretty straightforward, getters all have a special logic built in which tries to convert data from any reasonable source type to the requested type. If that cannot be done, an exception is raised.

For example, getter for the `AsString` property is called `CastToString` and in turn calls `TryCastToString`, which is in turn a public function of `TOmniValue`.

```
function TOmniValue.CastToString: string;
begin
  if not TryCastToString(Result) then
    raise Exception.Create('TOmniValue cannot be converted to string');
end;

function TOmniValue.TryCastToString(var value: string): boolean;
begin
  Result := true;
  case ovType of
    ovtNull:       value := '';
    ovtBoolean:    value := BoolToStr(AsBoolean, true);
    ovtInteger:    value := IntToStr(ovData);
    ovtDouble,
    ovtDateTime,
    ovtExtended:   value := FloatToStr(AsExtended);
    ovtAnsiString: value := string((ovIntf as IOmniAnsiStringData).Value);
    ovtString:     value := (ovIntf as IOmniStringData).Value;
    ovtWideString: value := (ovIntf as IOmniWideStringData).Value;
    ovtVariant:    value := string(AsVariant);
    else Result := false;
  end;
end;
```

When you don't know the data type stored in a `TOmniValue` variable and you don't want to raise an exception if compatible data is not available, you can use the `TryCastToXXX` family of functions directly.

```
function  TryCastToAnsiString(var value: AnsiString): boolean;
function  TryCastToBoolean(var value: boolean): boolean;
function  TryCastToCardinal(var value: cardinal): boolean;
function  TryCastToDouble(var value: Double): boolean;
function  TryCastToDateTime(var value: TDateTime): boolean;
function  TryCastToException(var value: Exception): boolean;
function  TryCastToExtended(var value: Extended): boolean;
function  TryCastToInt64(var value: int64): boolean;
function  TryCastToInteger(var value: integer): boolean;
function  TryCastToInterface(var value: IInterface): boolean;
function  TryCastToObject(var value: TObject): boolean;
function  TryCastToPointer(var value: pointer): boolean;
function  TryCastToString(var value: string): boolean;
function  TryCastToVariant(var value: Variant): boolean;
function  TryCastToWideString(var value: WideString): boolean;
```

Alternatively, you can use `CastToXXXDef` functions which return a default value if current value of the `TOmniValue` cannot be converted into required data type.

```
function  CastToAnsiStringDef(const defValue: AnsiString): AnsiString;
function  CastToBooleanDef(defValue: boolean): boolean;
function  CastToCardinalDef(defValue: cardinal): cardinal;
function  CastToDoubleDef(defValue: Double): Double;
function  CastToDateTimeDef(defValue: TDateTime): TDateTime;
function  CastToExceptionDef(defValue: Exception): Exception;
function  CastToExtendedDef(defValue: Extended): Extended;
function  CastToInt64Def(defValue: int64): int64;
function  CastToIntegerDef(defValue: integer): integer;
function  CastToInterfaceDef(const defValue: IInterface): IInterface;
function  CastToObjectDef(defValue: TObject): TObject;
function  CastToPointerDef(defValue: pointer): pointer;
function  CastToStringDef(const defValue: string): string;
function  CastToVariantDef(defValue: Variant): Variant;
function  CastToWideStringDef(defValue: WideString): WideString;
```

They are all implemented in the same value, similar to the `CastToObjectDef` below.

```
function TOmniValue.CastToObjectDef(defValue: TObject): TObject;
begin
  if not TryCastToObject(Result) then
    Result := defValue;
end;
```

1.7.2 Type Testing

For situations where you would like to determine the type of data stored inside the `TOmniValue`, there is the `IsXXX` family of functions.

```
function  IsAnsiString: boolean;
function  IsArray: boolean;
function  IsBoolean: boolean;
function  IsEmpty: boolean;
function  IsException: boolean;
function  IsFloating: boolean;
function  IsDateTime: boolean;
function  IsInteger: boolean;
function  IsInterface: boolean;
function  IsInterfacedType: boolean;
function  IsObject: boolean;
function  IsOwnedObject: boolean;
```

```
13  function  IsPointer: boolean;
14  function  IsRecord: boolean;
15  function  IsString: boolean;
16  function  IsVariant: boolean;
17  function  IsWideString: boolean;
```

Alternatively, you can use the `DataType` property.

```
1  type
2    TOmniValueDataType = (ovtNull, ovtBoolean, ovtInteger, ovtDouble, ovtObject,
3      ovtPointer, ovtDateTime, ovtException, ovtExtended, ovtString, ovtInterface,
4      ovtVariant, ovtWideString, ovtArray, ovtRecord, ovtAnsiString, ovtOwnedObject);
5
6  property DataType: TOmniValueDataType;
```

1.7.3 Clearing the Content

There are two ways to clear a `TOmniValue` – you can either call its `Clear` method, or you can assign it a `TOmniValue.Null`.

```
1  procedure Clear;
2  class function Null: TOmniValue; static;
```

An example:

```
1  var
2    ov: TOmniValue;
3
4  ov.Clear;
5  // or
6  ov := TOmniValue.Null;
```

1.7.4 Operators

`TOmniValue` also implements several `Implicit` operators which help with automatic conversion to and from different data types. Internally, they are implemented as assignment to/from the `AsXXX` property.

```
class operator Equal(const a: TOmniValue; i: integer): boolean;
class operator Equal(const a: TOmniValue; const s: string): boolean;
class operator Implicit(const a: AnsiString): TOmniValue;
class operator Implicit(const a: boolean): TOmniValue;
class operator Implicit(const a: Double): TOmniValue;
class operator Implicit(const a: Extended): TOmniValue;
class operator Implicit(const a: integer): TOmniValue;
class operator Implicit(const a: int64): TOmniValue;
class operator Implicit(const a: pointer): TOmniValue;
class operator Implicit(const a: string): TOmniValue;
class operator Implicit(const a: IInterface): TOmniValue;
class operator Implicit(const a: TObject): TOmniValue;
class operator Implicit(const a: Exception): TOmniValue;
class operator Implicit(const a: TOmniValue): AnsiString;
class operator Implicit(const a: TOmniValue): int64;
class operator Implicit(const a: TOmniValue): TObject;
class operator Implicit(const a: TOmniValue): Double;
class operator Implicit(const a: TOmniValue): Exception;
class operator Implicit(const a: TOmniValue): Extended;
class operator Implicit(const a: TOmniValue): string;
class operator Implicit(const a: TOmniValue): integer;
class operator Implicit(const a: TOmniValue): pointer;
class operator Implicit(const a: TOmniValue): WideString;
class operator Implicit(const a: TOmniValue): boolean;
class operator Implicit(const a: TOmniValue): IInterface;
class operator Implicit(const a: WideString): TOmniValue;
class operator Implicit(const a: Variant): TOmniValue;
class operator Implicit(const a: TDateTime): TOmniValue;
class operator Implicit(const a: TOmniValue): TDateTime;
```

Implicit conversion to/from `TDateTime` is supported only in Delphi XE and newer.

Two `Equal` operators simplify comparing `TOmniValue` to an `integer` and `string` data.

1.7.5 Use in Generic Types

Few methods simplify using `TOmniValue` in a generic class/record.

```
class function CastFrom<T>(const value: T): TOmniValue; static;
function  CastTo<T>: T;
function  CastToObject<T: class>: T; overload;
function  ToObject<T: class>: T;
```

`CastFrom<T>` converts any type into a `TOmniValue`. In Delphi 2009, this function is severely limited as only very simple types (integer, object) are supported. In Delphi 2010 and newer, `TValue` type is used to facilitate the conversion and all data types supported by the `TOmniValue` can be converted.

`CastTo<T>` converts `TOmniValue` into any other type. In Delphi 2009 same limitations apply as for `CastFrom<T>`.

`CastToObject<T>` (available in Delphi 2010 and newer) performs a *hard cast* with no type checking. It is equivalent to using `T(AsObject)`

`ToObject<T>` (available in Delphi 2010 and newer) casts the object to type `T` with type checking. It is equivalent to using `AsObject as T`.

1.7.6 Array Access

Each `TOmniValue` can contain an array of other `TOmniValues`. Internally, they are stored in a TOmniValueContainer object. This object can be accessed directly by reading the `AsArray` property.

```
1  property AsArray: TOmniValueContainer read GetAsArray;
```

IsArray can be used to test whether a `TOmniValue` contains an array of values.

Arrays can be accessed by an integer indexes (starting with `0`), or by string indexes (*named* access). Integer-indexed arrays are created by calling `TOmniValue.Create` and string-indexed arrays are created by calling `TOmniValue.CreateNamed`.

```
1  constructor Create(const values: array of const);
2  constructor CreateNamed(const values: array of const);
```

In the latter case, elements of the `values` parameter must alternate between names (string indexes) and values.

```
1  ov := TOmniValue.CreateNamed(
2          ['Key1', 'Value of ov[''Key1'']',
3           'Key2', 'Value of ov[''Key2'']'
4          ]);
```

In the example above, both `ov[0]` and `ov['Key1']` would return the same string, namely `'Value of ov[''Key1'']'`.

Array elements can be accessed with the `AsArrayItem` property, by using an integer index (for integer-indexed arrays), a string index (for string-indexed arrays), or a `TOmniValue` index. In the last case, the type of data stored inside the `TOmniValue` index parameter will determine how the array element is accessed. This last form is not available in Delphi 2007, where `AsArrayItemOV` should be used instead.

All forms of `AsArrayItem` allow extending an array. If you write data into an index which doesn't already exist, the array will automatically grow to accomodate the new value.

```
property AsArrayItem[idx: integer]: TOmniValue; default;
property AsArrayItem[const name: string]: TOmniValue; default;
property AsArrayItem[const param: TOmniValue]: TOmniValue; default;
property AsArrayItemOV[const param: TOmniValue]: TOmniValue;
```

If you want to test whether an array element exists, use the `HasArrayItem` function.

```
function  HasArrayItem(idx: integer): boolean; overload;
function  HasArrayItem(const name: string): boolean; overload;
function  HasArrayItem(const param: TOmniValue): boolean; overload;
```

In Delphi 2010 and newer `TOmniValue` also implements functions for converting data to and from `TArray<T>` for any supported type. `CastFrom<T>` and `CastTo<T>` functions are used internally to do the conversion.

```
class function FromArray<T>(const values: TArray<T>): TOmniValue; static;
function  ToArray<T>: TArray<T>;
```

1.7.7 Handling Records

A record `T` can be stored inside a `TOmniValue` by calling the `FromRecord<T>` function. To extract the data back into a record, use the `ToRecord<T>` function.

```
class function FromRecord<T: record>(const value: T): TOmniValue; static;
function  ToRecord<T>: T;
```

An example:

```
var
  ts: TTimeStamp;
  ov: TOmniValue;

ov := TOmniValue.FromRecord<TTimeStamp>(ts);
ts := ov.ToRecord<TTimeStamp>;
```

`TOmniValue` jumps through quite some hoops to store a record. It is first converted into a TOmniRecordWrapper which is then wrapped inside an IOmniAutoDestroyObject interface to provide a reference-counted lifetime management.

Because of all that storing records inside `TOmniValue` is, although very flexible, not that fast.

```
class function TOmniValue.FromRecord<T>(const value: T): TOmniValue;
begin
  Result.SetAsRecord(
    CreateAutoDestroyObject(
      TOmniRecordWrapper<T>.Create(value)));
end;
```

1.7.8 Object Ownership

`TOmniValue` can take an ownership of a `TObject`-type data. To achieve that, you can either assign an object to the `AsOwnedObject` property or set the `OwnsObject` property to `True`.

```
property  AsOwnedObject: TObject;
function  IsOwnedObject: boolean;
property OwnsObject: boolean;
```

When a object-owning `TOmniValue` goes out of scope, the owned object is automatically destroyed.

You can change the ownership status at any time by assigning to the `OwnsObject` property.

1.7.9 Working With TValue

In Delphi 2010 and newer, `TOmniValue` also provides a `AsTValue` property and corresponding `Implicit` operator so you can easily convert a `TValue` data into `TOmniValue` and back.

```
class operator Implicit(const a: TValue): TOmniValue; inline;
class operator Implicit(const a: TOmniValue): TValue; inline;
property AsTValue: TValue;
```

1.7.10 Low-level Methods

For people with special needs (and for internal OmniThreadLibrary use), `TOmniValue` exposes following public methods.

```
procedure _AddRef;
procedure _Release;
procedure _ReleaseAndClear;
function  RawData: PInt64;
procedure RawZero;
```

`_AddRef` increments reference counter of stored data if `TOmniValue` contains such data.

`_Release` decrements reference counter of stored data if `TOmniValue` contains such data.

`_ReleaseAndClear` is just a shorthand for calling a `_Release` followed by a call to `RawZero`.

`RawData` returns pointer to the data stored in the `TOmniValue`.

`RawZero` clears the stored data without decrementing the reference counter.

1.8 TOmniValueObj

The *OtlCommon* unit also implements a simple object which can wrap a `TOmniValue` for situations where you would like to store it inside a data structure that only supports object types.

```
1  TOmniValueObj = class
2    constructor Create(const value: TOmniValue);
3    property Value: TOmniValue read FValue;
4  end;
```

1.9 Fluent Interfaces

OmniThreadLibrary heavily uses fluent interface[8] approach. Most of functions in OmniThreadLibrary interfaces are returning `Self` as the output value. Take for example this declaration of the pipeline abstraction, slightly edited for brevity.

```
1  IOmniPipeline = interface
2    procedure Cancel;
3    function  From(const queue: IOmniBlockingCollection): IOmniPipeline;
4    function  HandleExceptions: IOmniPipeline;
5    function  NumTasks(numTasks: integer): IOmniPipeline;
6    function  OnStop(const stopCode: TProc): IOmniPipeline;
7    function  Run: IOmniPipeline;
8    function  Stage(
9      pipelineStage: TPipelineSimpleStageDelegate;
10     taskConfig: IOmniTaskConfig = nil): IOmniPipeline; overload;
11   function  Stage(
12     pipelineStage: TPipelineStageDelegate;
13     taskConfig: IOmniTaskConfig = nil): IOmniPipeline; overload;
14   function  Stage(
15     pipelineStage: TPipelineStageDelegateEx;
16     taskConfig: IOmniTaskConfig = nil): IOmniPipeline; overload;
17   function  Stages(
18     const pipelineStages: array of TPipelineSimpleStageDelegate;
19     taskConfig: IOmniTaskConfig = nil): IOmniPipeline; overload;
20   function  Stages(
21     const pipelineStages: array of TPipelineStageDelegate;
22     taskConfig: IOmniTaskConfig = nil): IOmniPipeline; overload;
23   function  Stages(
24     const pipelineStages: array of TPipelineStageDelegateEx;
25     taskConfig: IOmniTaskConfig = nil): IOmniPipeline; overload;
26   function  Throttle(numEntries: integer; unblockAtCount: integer = 0):
```

[8] http://en.wikipedia.org/wiki/Fluent_interface

```
27      IOmniPipeline;
28    function  WaitFor(timeout_ms: cardinal): boolean;
29    property Input: IOmniBlockingCollection read GetInput;
30    property Output: IOmniBlockingCollection read GetOutput;
31  end;
```

As you can see, most of the functions return the `IOmniPipeline` interface. In code, this is implemented by returning `Self`.

```
1   function TOmniPipeline.From(
2     const queue: IOmniBlockingCollection): IOmniPipeline;
3   begin
4     opInput := queue;
5     Result := Self;
6   end;
```

This allows calls to such interfaces to be *chained*. For example, the following code from the Pipeline section shows how to use `Parallel.Pipeline` without ever storing the resulting interface in a variable.

```
1   var
2     sum: integer;
3
4   sum := Parallel.Pipeline
5     .Stage(
6       procedure (const input, output: IOmniBlockingCollection)
7       var
8         i: integer;
9       begin
10        for i := 1 to 1000000 do
11          output.Add(i);
12      end)
13    .Stage(
14      procedure (const input: TOmniValue; var output: TOmniValue)
15      begin
16        output := input.AsInteger * 3;
17      end)
18    .Stage(
19      procedure (const input, output: IOmniBlockingCollection)
20      var
21        sum: integer;
22        value: TOmniValue;
23      begin
24        sum := 0;
25        for value in input do
```

```
26          Inc(sum, value);
27        output.Add(sum);
28      end)
29    .Run.Output.Next;
```

If you don't like fluent interface approach, don't worry. OmniThreadLibrary can be used without it. You can always call a function as if it is a procedure and compiler will just throw away the result.

The example above could be rewritten as such.

```
1  var
2    sum: integer;
3    pipe: IOmniPipeline;
4
5  pipe := Parallel.Pipeline;
6  pipe.Stage(
7    procedure (const input, output: IOmniBlockingCollection)
8    var
9      i: integer;
10   begin
11     for i := 1 to 1000000 do
12       output.Add(i);
13   end);
14 pipe.Stage(
15   procedure (const input: TOmniValue; var output: TOmniValue)
16   begin
17     output := input.AsInteger * 3;
18   end);
19 pipe.Stage(
20   procedure (const input, output: IOmniBlockingCollection)
21   var
22     sum: integer;
23     value: TOmniValue;
24   begin
25     sum := 0;
26     for value in input do
27       Inc(sum, value);
28     output.Add(sum);
29   end);
30 pipe.Run;
31 sum := pipe.Output.Next;
```

2. High-level Multithreading

Face it – multithreading programming is hard. It is hard to design a multithread program, it is hard to write and test it and it is *insanely* hard to debug it. To alleviate this problem, OmniThreadLibrary introduces a number of pre-packaged multithreading solutions; so-called *abstractions*.

The idea behind the high-level abstractions is that the user should just choose appropriate abstraction and write the worker code, while the OmniThreadLibrary provides the framework that implements the tricky multithreaded parts, takes care of synchronisation and so on.

2.1 Async

Async is the simplest of high-level abstractions and is typically used for fire and forget scenarios. To create an Async task, call Parallel.Async.



When you call `Parallel.Async`, code is started in a new thread (indicated by the bold vertical line) and both main and background threads continue execution. At some time, background task completes execution and disappears.

See also demo `46_Async`.

Example:

```
Parallel.Async(
  procedure
  begin
    MessageBeep($FFFFFFFF);
  end);
```

This simple program creates a background task with a sole purpose to make some noise from it. The task is coded as an anonymous method but you can also use a normal method or a normal procedure for the task code.

The `Parallel` class defines two `Async` overloads. The first accepts a parameter-less background task and an optional task configuration block and the second accepts a background task with an IOmniTask parameter and an optional task configuration block.

```
type
  TOmniTaskDelegate = reference to procedure(const task: IOmniTask);

  Parallel = class
    class procedure Async(task: TProc;
      taskConfig: IOmniTaskConfig = nil); overload;
    class procedure Async(task: TOmniTaskDelegate;
      taskConfig: IOmniTaskConfig = nil); overload;
    ...
  end;
```

The second form is useful if the background code needs access to the IOmniTask interface, for example to send messages to the owner or to execute code in the owner thread (typically that will be the main thread).

The example below uses Async task to fetch the contents of a web page (by calling a mysterious function HttpGet) and then uses Invoke to execute a code that logs the length of the result in the main thread.

```
Parallel.Async(
  procedure (const task: IOmniTask)
  var
    page: string;
  begin
    HttpGet('otl.17slon.com', 80, 'tutorials.htm', page, '');
    task.Invoke(
      procedure
      begin
        lbLogAsync.Items.Add(Format('Async GET: %d ms; page length = %d',
          [time, Length(page)]))
      end);
  end);
```

The same result could be achieved by sending a message from the background thread to the main thread. TaskConfig block is used to configure message handler.

```
const
  WM_RESULT = WM_USER;

procedure LogResult(const task: IOmniTaskControl; const msg: TOmniMessage);
begin
  lbLogAsync.Items.Add(Format('Async GET: %d ms; page length = %d',
    [time, Length(page)]))
end;

Parallel.Async(
  procedure (const task: IOmniTask)
  var
    page: string;
  begin
    HttpGet('otl.17slon.com', 80, 'tutorials.htm', page, '');
    task.Comm.Send(WM_RESULT, page);
  end,
  TaskConfig.OnMessage(WM_RESULT, LogResult)
);
```

Let me warn you that in cases where you want to return a result from a background task, Async abstraction is not the most appropriate. You would be better off by using a Future.

2.1.1 Handling Exceptions

If the background code raises unhandled exception, OmniThreadLibrary will catch this exception and re-raise it in the `OnTerminated` handler. This way the exception will travel from the background thread to the owner thread where it can be processed.

As the `OnTerminated` handler occurs at the unspecified moment when Windows are processing window messages, there is no good way to catch this message with a try..except block. The caller must install its own `OnTerminated` handler instead and handle exception there.

The following example uses `OnTerminated` handler to detach fatal exception from the task, log the exception details and destroy the exception object.

```
Parallel.Async(
  procedure
  begin
    Sleep(1000);
    raise Exception.Create('Exception in Async');
  end,
  Parallel.TaskConfig.OnTerminated(
    procedure (const task: IOmniTaskControl)
    var
      excp: Exception;
    begin
      if assigned(task.FatalException) then begin
        excp := task.DetachException;
        Log('Caught async exception %s:%s',[excp.ClassName, excp.Message]);
        FreeAndNil(excp);
      end;
    end
  ));
```

If you don't install a `OnTerminated` handler, exception will be handled by the application-level filter, which will by default cause a message box to appear.

See also demo `48_OtlParallelExceptions`.

2.2 Async/Await

Async/Await is a simplified version of the Async abstraction which mimics the .NET Async/Await[1] mechanism.[2]

In short, Async/Await accepts two parameterless anonymous methods. The first one is executed in a background thread and the second one is executed in the main thread after the background thread has completed its work.

See also demo `53_AsyncAwait`.

Using Async/Await you can, for example, create a background operation which is triggered by a click and which re-enables button after the background job has been completed.

[1] http://blogs.msdn.com/b/pfxteam/archive/2012/04/12/10293335.aspx
[2] Read more at http://www.thedelphigeek.com/2012/07/asyncawait-in-delphi.html.

```
procedure TForm1.Button1Click(Sender: TObject);
var
  button: TButton;
begin
  button := Sender as TButton;
  button.Caption := 'Working ...';
  button.Enabled := false;
  Async(
    // executed in a background thread
    procedure begin
      Sleep(5000);
    end).
  Await(
    // executed in the main thread after
    // the anonymous method passed to
    // Async has completed its work
    procedure begin
      button.Enabled := true;
      button.Caption := 'Done!';
    end);
end;
```

Keep in mind that Async is invoked by calling `Parallel.Async` and Async/Await by calling `Async`.

Exceptions in the `Async` part are currently not handled by the OmniThreadLibrary.

2.3 For Each

For Each abstraction creates a parallel *for* loop that iterates over a range of data (number range, list, queue, dataset ...) in multiple threads. To create a parallel for loop, call `Parallel.ForEach`.



When you use Parallel.ForEach, OmniThreadLibrary starts multiple background tasks and connects them to the source through a serialization mechanism. Output is optionally sorted in the order of the input. By default, ForEach waits for all background threads to complete before the control is returned to the caller.

See also demos `35_ParallelFor` and `36_OrderedFor`.

OmmiThreadLibrary also implements For abstraction, which is more limited than *ForEach* but executes faster.

Example:

```
1  PrimeCount.Value := 0;
2  Parallel.ForEach(1, 1000000).Execute(
3    procedure (const value: integer)
4    begin
5      if IsPrime(value) then
6        PrimeCount.Increment;
7      end;
8    end);
```

This simple program calculates number of prime numbers in the range from one to one million. The `PrimeCount` object must be capable of atomic increment (a thread-safe increment), which is simple to achieve with the use of the TGp4AlignedInt record. The ForEach task is coded as an anonymous method but you can also use a normal method or a normal procedure for the task code.

2.3.1 Cooperation

The main point of the `ForEach` abstraction is cooperation between the parallel tasks. `ForEach` goes to great lengths to minimize possible clashes between threads when they access the source data. Except in special occasions (number range, IOmniBlockingCollection), source data is not thread-safe and locking must be used for access.

To minimize this locking, source data is allocated to worker tasks in blocks. `ForEach` creates a *source provider* object which will access the source data in a thread-safe manner. This source provider makes sure to always return an appropriately-sized block of source data (size will depend on number of tasks, type of the source data and other factors) when a task runs out of data to process.

Because the source data is allocated in blocks, it is possible that one of the tasks runs out of work while other tasks are still busy. In this case, a task will *steal* data from one of the other tasks. This approach makes all tasks as busy as possible while minimizing the contention.

The details of this process are further discussed in section Internals below.

2.3.2 Iterating over ...

The `Parallel` class defines many `ForEach` overloads, each supporting different container type. We will look at them in more detail in the following sections.

2.3.2.1 ... Number Ranges

To iterate over a range, pass `first` and `last` index to the `ForEach` call. Optionally, you can pass a `step` parameter, which defaults to `1`. `ForEach` will then iterate from `first` to `last` with a `step` increment.

```
class function ForEach(low, high: integer; step: integer = 1):
  IOmniParallelLoop<integer>; overload;
```

The pseudocode for *numeric* `ForEach` could be written as

```
i := low;
while ((step > 0) and (i <= high)) or
      ((step < 0) and (i >= high)) do
begin
  // process 'i' in parallel
  if low < high then Inc(i, step)
                else Dec(i, step);
end;
```

2.3.2.2 ... Enumerable Collections

If you want to iterate over a collection (say, a `TStringList`), you have two possibilities.

One is to use an equivalent of `for i := 0 to sl.Count-1 do Something(sl[i])`.

```
Parallel.ForEach(0, sl.Count-1).Execute(
  procedure (const value: integer)
  begin
    Something(sl[value]);
  end);
```

Another is to use an equivalent of `for s in sl do Something(s)`.

```
Parallel.ForEach(sl).Execute(
  procedure (const value: TOmniValue)
  begin
    Something(value);
  end);
```

In the second example, value is passed to the task function as a `TOmniValue` parameter. In the example above, it will be automatically converted into a string, but sometimes you'll have to do it manually, by calling `value.AsString` (or use other appropriate casting function when iterating over a different container).

A variation of the second approach is to tell the `ForEach` that the container contains strings. OmniThreadLibrary will then do the conversion for you.

```
Parallel.ForEach<string>(sl).Execute(
  procedure (const value: string)
  begin
    Something(value);
  end);
```

You may wonder which of those approaches is better. The answer depends on whether you can simultaneously access different items in the container from different threads at the same time. In other words, you have to know whether the container is thread-safe for reading. Luckily, all important Delphi containers (`TList`, `TObjectList`, `TStringList`) fall into this category.

If the container is thread-safe for reading, then the *numeric* approach (`ForEach(0, sl.Count-1)`) is **much** faster than the *for..in* approach (`ForEach(sl)`). The speed difference comes from the locking - in the former example `ForEach` never locks anything and in the latter example locking is used to synchronize access to the container.

However, if the container is not thread-safe for reading, you **have** to use the latter approach.

There are three ways to iterate over enumerable containers. You can provide the `ForEach` call with an `IEnumerable` interface, with an `IEnumerator` interface or with an enumerable collection itself. In the latter case, OmniThreadLibrary will use RTTI to access the enumerator for the collection. For this to work, enumerator itself must be implemented as an object, not as a record or interface. Luckily, most if not all of the VCL enumerators are implemented in this way.

```
class function  ForEach(const enumerable: IEnumerable):
  IOmniParallelLoop; overload;
class function  ForEach(const enum: IEnumerator):
  IOmniParallelLoop; overload;
class function  ForEach(const enumerable: TObject):
  IOmniParallelLoop; overload;
class function  ForEach<T>(const enumerable: IEnumerable):
  IOmniParallelLoop<T>; overload;
class function  ForEach<T>(const enum: IEnumerator):
  IOmniParallelLoop<T>; overload;
class function  ForEach<T>(const enumerable: TEnumerable<T>):
  IOmniParallelLoop<T>; overload;
class function  ForEach<T>(const enum: TEnumerator<T>):
  IOmniParallelLoop<T>; overload;
class function  ForEach<T>(const enumerable: TObject):
  IOmniParallelLoop<T>; overload;
```

2.3.2.3 ... Thread-safe Enumerable Collections

Collection enumeration uses locking to synchronize access to the collection enumerator, which slows down the enumeration process. In some special cases, collection may be enumerable without the locking. To enumerate over such collection, it must implement `IOmniValueEnumerable` and `IOmniValueEnumerator` interfaces, which are defined in the OtlCommon unit.

```
class function  ForEach(const enumerable: IOmniValueEnumerable):
  IOmniParallelLoop; overload;
class function  ForEach(const enum: IOmniValueEnumerator):
  IOmniParallelLoop; overload;
class function  ForEach<T>(const enumerable: IOmniValueEnumerable):
  IOmniParallelLoop<T>; overload;
class function  ForEach<T>(const enum: IOmniValueEnumerator):
  IOmniParallelLoop<T>; overload;
```

2.3.2.4 ... Blocking Collections

To simplify enumerating over blocking collections, the `Parallel` class implements two `ForEach` overloads accepting a blocking collection. Internally, blocking collection is enumerated with the `IOmniValueEnumerable` interface.

```
class function  ForEach(const source: IOmniBlockingCollection):
  IOmniParallelLoop; overload;
class function  ForEach<T>(const source: IOmniBlockingCollection):
  IOmniParallelLoop<T>; overload;
```

2.3.2.5 ... Anything

As a last resort, the `Parallel` class implements three `ForEach` overloads that will (with some help from the programmer) iterate over any data.

The `TOmniSourceProvider` way is powerful, but complicated.

```
class function  ForEach(const sourceProvider: TOmniSourceProvider):
  IOmniParallelLoop; overload;
```

You must implement a descendant of the `TOmniSourceProvider` class. All methods must be thread-safe. For more information about the source providers, see the Internals section, below.

```
TOmniSourceProvider = class abstract
public
  function  Count: int64; virtual; abstract;
  function  CreateDataPackage: TOmniDataPackage; virtual; abstract;
  function  GetCapabilities: TOmniSourceProviderCapabilities;
    virtual; abstract;
  function  GetPackage(dataCount: integer; package: TOmniDataPackage):
    boolean; virtual; abstract;
  function  GetPackageSizeLimit: integer; virtual; abstract;
end;
```

As this approach is not for the faint of heart, OmniThreadLibrary provides a slower but much simpler version.

```
class function  ForEach(enumerator: TEnumeratorDelegate):
  IOmniParallelLoop; overload;
class function  ForEach<T>(enumerator: TEnumeratorDelegate<T>):
  IOmniParallelLoop<T>; overload;
```

Here, you must provide a function that will return *next* data whenever the `ForEach` asks for it.

```
TEnumeratorDelegate = reference to function(var next: TOmniValue): boolean;
TEnumeratorDelegate<T> = reference to function(var next: T): boolean;
```

OmniThreadLibrary will provide the synchronisation (locking) so you can be sure this method will only be called from one thread at any time. As you may expect, this will slow things down, but parallelization may still give you a reasonable performance increase if `ForEach` payload is substantial (i.e. if the method you are executing in the `ForEach` loop takes some time to execute).

The `TEnumeratorDelegate` function can also be used as a generator; that is it can *calculate* the values that will then be processed in the parallel for loop.

2.3.3 Providing External Input

Sometimes, especially when you are dealing with datasets, synchronized access to the container will not be enough. When you are dealing with database connections, datasets etc you can easily run into *thread affinity* problems - that is the inability of some component to work correctly if it is called from a different thread than the one that it was created in.

> [Always initialize database connections and datasets in the thread that will use them. You code *may* work without that precaution but unless you have extensively tested database components in multiple threads, you should not assume that they will work correctly unless that condition (initialization and use in the same thread) is met.]

In such case, the best way is to provide the input directly from the main thread. There are few different ways to achieve that.

1. Repackage data into another collection that can be easily consumed in `ForEach` (`TObjectList`, `TStringList`, `TOmniBlockingCollection`).
2. Run the `ForEach` in `NoWait` mode, then write the data into the input queue and when you run out of data, wait for the `ForEach` loop to terminate. This approach is also useful, when you want to push `ForEach` into background and provide it with data from some asynchronous event handler.

An example of the second approach will help clarify the idea.

```
uses
  OtlCommon,
  OtlCollections,
  OtlParallel;

procedure Test;
var
  i    : integer;
  input: IOmniBlockingCollection;
  loop : IOmniParallelLoop<integer>;
  wait : IOmniWaitableValue;
begin
  // create the container
  input := TOmniBlockingCollection.Create;
  // create the 'end of work' signal
  wait := CreateWaitableValue;
  loop := Parallel.ForEach<integer>(input);
  // set up the termination method which will signal 'end of work'
  loop.OnStop(
    procedure
    begin
      wait.Signal;
    end);
  // start the parallel for loop in NoWait mode
  loop.NoWait.Execute(
    procedure (const value: integer)
    begin
      // do something with the input value
      OutputDebugString(PChar(Format('%d', [value])));
    end
  );
  // provide the data to the parallel for loop
  for i := 1 to 1000 do
    input.Add(i);
  // signal to the parallel for loop that there's no more data to process
  input.CompleteAdding;
  // wait for the parallel for loop to stop
  wait.WaitFor;
  // destroy the parallel for loop
  loop := nil;
end;
```

2.3.4 IOmniParallelLoop Interface

The `Parallel.ForEach` returns an `IOmniParallelLoop` interface which is used to configure and run the parallel for loop.

```
IOmniParallelLoop = interface
  function  Aggregate(defaultAggregateValue: TOmniValue;
    aggregator: TOmniAggregatorDelegate): IOmniParallelAggregatorLoop;
  function  AggregateSum: IOmniParallelAggregatorLoop;
  procedure Execute(loopBody: TOmniIteratorDelegate); overload;
  procedure Execute(loopBody: TOmniIteratorTaskDelegate); overload;
  function  CancelWith(const token: IOmniCancellationToken):
    IOmniParallelLoop;
  function  Initialize(taskInitializer: TOmniTaskInitializerDelegate):
    IOmniParallelInitializedLoop;
  function  Into(const queue: IOmniBlockingCollection):
    IOmniParallelIntoLoop; overload;
  function  NoWait: IOmniParallelLoop;
  function  NumTasks(taskCount : integer): IOmniParallelLoop;
  function  OnMessage(eventDispatcher: TObject):
    IOmniParallelLoop; overload; deprecated 'use TaskConfig';
  function  OnMessage(msgID: word; eventHandler: TOmniTaskMessageEvent):
    IOmniParallelLoop; overload; deprecated 'use TaskConfig';
  function  OnMessage(msgID: word; eventHandler: TOmniOnMessageFunction):
    IOmniParallelLoop; overload; deprecated 'use TaskConfig';
  function  OnTaskCreate(taskCreateDelegate: TOmniTaskCreateDelegate):
    IOmniParallelLoop; overload;
  function  OnTaskCreate(taskCreateDelegate:
    TOmniTaskControlCreateDelegate): IOmniParallelLoop; overload;
  function  OnStop(stopCode: TProc): IOmniParallelLoop;
  function  PreserveOrder: IOmniParallelLoop;
  function  TaskConfig(const config: IOmniTaskConfig): IOmniParallelLoop;
end;
```

`ForEach<T>` returns an `IOmniParallelLoop<T>` interface, which is exactly the same as the `IOmniParallelLoop` except that each method returns the appropriate `<T>` version of the interface.

`Aggregate` and `AggregateSum` are used to implement aggregation. See the Aggregation section, below.

`Execute` accepts the block of code to be executed for each value in the input container. Two method signatures are supported, both with the `<T>` variant. One accepts only the iteration value and another accepts the `IOmniTask` parameter.

```
TOmniIteratorDelegate = reference to procedure(const value: TOmniValue);
TOmniIteratorDelegate<T> = reference to procedure(const value: T);
TOmniIteratorTaskDelegate =
  reference to procedure(const task: IOmniTask; const value: TOmniValue);
TOmniIteratorTaskDelegate<T> =
  reference to procedure(const task: IOmniTask; const value: T);
```

`CancelWith` enables the cancellation mechanism.

With `Initialize` and `OnTaskCreate` you can initialize per-task data before the task begins execution. See the Task Initialization section, below.

`Into` sets up the output queue, see Preserving Output Order.

If you call the `NoWait` function, parallel for will start in the background and control will be returned to the main thread immediately. If `NoWait` is not called, `Execute` will only return after all tasks have stopped working.

By calling `NumTasks` you can set up the number of worker tasks. By default, number of tasks is set to *[number of cores available to the process] - 1* if `NoWait` or `PreserveOrder` modifiers are used and to *[number of cores available to the process]* in all other cases.

If `NumTasks` receives a positive parameter (`> 0`), then the number of worker tasks is set to that number. For example, `NumTasks(16)` starts 16 worker tasks, even if that is more then number of available cores.

If `NumTasks` receives a negative parameter (`< 0`), it specifies number of cores that should be reserved for other use. Number of worker task is then set to `<number of available cores> - <number of reserved cores>`. If, for example, current process can use 16 cores and `NumTasks(-4)` is used, only `12` (`16-4`) worker tasks will be started.

Parameter `0` is not allowed and results in an exception.

`OnMessage` functions are deprecated, use `TaskConfig` instead.

`OnStop` sets up a termination handler which will be called after all parallel for tasks will have completed their work. If `NoWait` function was called, `OnStop` will be called from one of the worker threads. If, however, `NoWait` function was not called, `OnStop` will be called from the thread that created the `ForEach` abstraction. This behaviour makes it hard to execute VCL code from the `OnStop` so release 3.02 introduced another variation accepting a delegate with an `IOmniTask` parameter.

```
TOmniTaskStopDelegate = reference to procedure (const task: IOmniTask);
IOmniParallelLoop = interface
  function  OnStop(stopCode: TOmniTaskStopDelegate): IOmniParallelLoop;
    overload;
end;
```

Using this version of `OnStop`, the termination handler can use `task.Invoke` to execute some code in the main thread. This, however, requires the `ForEach` abstraction to stay alive until the `Invoke`-d code is executed so you must store the `ForEach` result in a global variable (form field, for example) and destroy it only in the termination handler.

```
var
  loop: IOmniParallelLoop<integer>;

loop := Parallel.ForEach(1, N).NoWait;
loop.OnStop(
  procedure (const task: IOmniTask)
  begin
    task.Invoke(
      procedure
      begin
        // do anything
        loop := nil;
      end);
  end);
loop.Execute(
  procedure (const value: integer)
  begin
    ...
  end);
```

`PreserveOrder` modifies the parallel for behaviour so that output values are generated in the order of the corresponding input values. See the Preserving Output Order section, below.

`TaskConfig` sets up a task configuration block. Same task configuration block will be applied to all worker tasks.

The following example uses `TaskConfig` to set up a message handler which will receive messages sent from `ForEach` worker tasks.

```
FParallel := Parallel.ForEach(1, 17)
  .TaskConfig(Parallel.TaskConfig.OnMessage(Self))
  .NoWait
  .OnStop(procedure begin FParallel := nil; end);

FParallel
  .Execute(
    procedure (const task: IOmniTask; const value: integer)
    begin
      task.Comm.Send(WM_LOG, value);
    end);
```

Messages sent from the worker task are received and dispatched by the `IOmniParallelLoop` interface. This requires the `ForEach` abstraction to stay alive until the messages are processed so you must store the `ForEach` result in a global variable (form field, for example) and destroy it only in the `OnStop` handler.

Some functions return a different interface. Typically, it only implements the `Execute` function accepting a different parameter than the 'normal' `Execute`. For example, `Aggregate` returns the `IOmniParallelAggregatorLoop` interface.

```
TOmniIteratorIntoDelegate =
  reference to procedure(const value: TOmniValue; var result: TOmniValue);

IOmniParallelAggregatorLoop = interface
  function  Execute(loopBody: TOmniIteratorIntoDelegate): TOmniValue;
end;
```

These variants of the `IOmniParallelLoop` interface will be described in following sections.

2.3.5 Preserving Output Order

When you run a `ForEach` loop, you can't tell in advance in which order elements from the input collection will be processed in. For example, the code below will generate all primes from `1` to `CMaxPrime` and write them into the output queue (`primeQueue`) in a nondeterministic order.

```
primeQueue := TOmniBlockingCollection.Create;
Parallel.ForEach(1, CMaxPrime).Execute(
  procedure (const value: integer)
  begin
    if IsPrime(value) then begin
      primeQueue.Add(value);
    end;
  end);
```

Sometimes this will represent a big problem and you'll have to write a sorting function that will resort the output before it can be processed further. To alleviate the problem, `IOmniParallelLoop` implements the `PreserveOrder` modifier. When used, `ForEach` loop will internally sort the results produced in the task method passed to the `Execute` method.

Using `PreserveOrder` also forces you to use the `Into` method which returns the `IOmniParallelIntoLoop` interface. (As you may expect, there's also the `<T>` version of that interface.)

```
TOmniIteratorIntoDelegate =
  reference to procedure(const value: TOmniValue; var result: TOmniValue);
TOmniIteratorIntoTaskDelegate =
  reference to procedure(const task: IOmniTask; const value: TOmniValue;
                         var result: TOmniValue);

IOmniParallelIntoLoop = interface
  procedure Execute(loopBody: TOmniIteratorIntoDelegate); overload;
  procedure Execute(loopBody: TOmniIteratorIntoTaskDelegate); overload;
end;
```

As you can see, the `Execute` method in `IOmniParallelIntoLoop` takes a different parameter than the 'normal' `Execute`. Because of that, you'll have to change a code that is passed to the `Execute` to return a result.

```
primeQueue := TOmniBlockingCollection.Create;
Parallel.ForEach(1, CMaxPrime)
  .PreserveOrder
  .Into(primeQueue)
  .Execute(
    procedure (const value: integer; var res: TOmniValue)
    begin
      if IsPrime(value) then
        res := value;
    end);
```

When using `PreserveOrder` and `Into`, `ForEach` calls your worker code for each input value. If the worker code sets output parameter (`res`) to any value, it will be inserted into a temporary buffer. Then the magic will happen (see the Internals section, below) and as soon as the appropriate (sorted) value is available in the temporary buffer, it is inserted into the output queue (the one passed to the `Into` parameter).

You can also use `Into` without the `PreserveOrder`. This will give you queue management but no ordering.

2.3.6 Aggregation

Aggregation allows you to collect data from the parallel for tasks and calculate one number that is returned to the user.

Let's start with an example - and a very bad one! The following code fragment tries to calculate a number of prime numbers between `1` and `CMaxPrime`.

```
numPrimes := 0;
Parallel.ForEach(1, CMaxPrime).Execute(
  procedure (const value: integer)
  begin
    if IsPrime(value) then
      Inc(numPrimes);
  end);
```

Let's say it out loud - this code is **wrong**! Access to the shared variable is not synchronized between threads and that will make the result indeterminable. One way to solve the problem is to wrap the `Inc(numPrimes)` with locking and another is to use `InterlockedIncrement` instead of `Inc`, but both will slow down the execution a lot.

A solution to this problem is to use the `Aggregate` function.

```
procedure SumPrimes(var aggregate: TOmniValue; const value: TOmniValue)
begin
  aggregate := aggregate.AsInt64 + value.AsInt64;
end;

procedure CheckPrime(const value: integer; var result: TOmniValue)
begin
  if IsPrime(value) then
    Result := 1;
end;

numPrimes :=
  Parallel.ForEach(1, CMaxPrime)
  .Aggregate(0, SumPrimes)
  .Execute(CheckPrime);
```

`Aggregate` takes two parameters - the first is the initial value for the aggregate and the second is an aggregation function - a piece of code that will take the current aggregate value and update it with the value returned from the parallel for task.

When using `Aggregate`, parallel for task (the code passed to the `Execute` function) has the same signature as when used with `Into`. It takes the current iteration value and optionally produces a result.

We could replace the code above with a `for` loop.

```
agg := 0;
result.Clear;
for value := 1 to CMaxPrime do begin
  CheckPrime(value, result);
  if not result.IsEmpty then begin
    SumPrimes(agg, result);
    result.Clear;
  end;
end;
numPrimes := agg;
```

`ForEach` executes the aggregation in two stages. While the parallel for task is running, it will use this approach to aggregate data into a local variable. When it runs out of work, it will call the same aggregation method to aggregate this local variable into a global result. In this second stage, however, locking will be used to protect the access to the global result.



Because summation is the most common usage of aggregation, `IOmniParallelLoop` implements function `AggregateSum`, which works exactly the same as the `SumPrimes` above.

```
numPrimes :=
  Parallel.ForEach(1, CMaxPrime)
   .AggregateSum
   .Execute(
     procedure (const value: integer; var result: TOmniValue)
     begin
       if IsPrime(value) then
         Result := 1;
     end
   );
```

Aggregation function can do something else but the summation. The following code segment uses aggregation to find the length of the longest line in a file.

```
function GetLongestLineInFile(const fileName: string): integer;
var
  maxLength: TOmniValue;
  sl       : TStringList;
begin
  sl := TStringList.Create;
  try
    sl.LoadFromFile(fileName);
    maxLength := Parallel.ForEach<string>(sl)
      .Aggregate(0,
        procedure(var aggregate: TOmniValue; const value: TOmniValue)
        begin
          if value.AsInteger > aggregate.AsInteger then
            aggregate := value.AsInteger;
        end)
      .Execute(
        procedure(const value: string; var result: TOmniValue)
        begin
          result := Length(value);
        end);
        Result := maxLength;
  finally FreeAndNil(sl); end;
end;
```

2.3.7 Cancellation

`ForEach` has a built-in cancellation mechanism. To use it, create a cancellation token and pass it to the `CancelWith` function. When a cancellation token gets signalled, all worker loops will complete the current iteration and then stop.

An example of using cancellation token can be found in the chapter Parallel Search in a Tree.

2.3.8 Task Initialization and Finalization

In some cases it would be nice if each parallel task could have some data initialized at beginning and available to the enumerator code (the one passed to the `Execute`). For those occasions, `ForEach` implements an `Initialize` function.

```
TOmniTaskInitializerDelegate =
  reference to procedure(var taskState: TOmniValue);
TOmniTaskFinalizerDelegate =
  reference to procedure(const taskState: TOmniValue);
TOmniIteratorStateDelegate =
  reference to procedure(const value: TOmniValue; var taskState: TOmniValue);

IOmniParallelInitializedLoop = interface
  function  Finalize(taskFinalizer: TOmniTaskFinalizerDelegate):
    IOmniParallelInitializedLoop;
  procedure Execute(loopBody: TOmniIteratorStateDelegate);
end;

IOmniParallelLoop = interface
  ...
  function  Initialize(taskInitializer: TOmniTaskInitializerDelegate):
    IOmniParallelInitializedLoop;
end;
```

You provide `Initialize` with *task initializer*, a procedure that will be called in each parallel worker task when it is created and before it starts enumerating values. This procedure can initialize the `taskState` parameter with any value.

`Initialize` returns a `IOmniParallelInitializedLoop` interface which implements two functions - `Finalize` and `Execute`. Call `Finalize` to set up *task finalizer*, a procedure that gets called after all values have been enumerated and before the parallel worker task ends its job.

`Execute` accepts a worker method with two parameters - the first one is the usual value from the enumerated container and the second contains the shared task state.

Of course, all those functions and interfaces are implemented in the `<T>` version, too.

The following example shows how to calculate number of primes from `1` to `CHighPrime` by using initializers and finalizers.

```
var
  lockNum  : TOmniCS;
  numPrimes: integer;
begin
  numPrimes := 0;
  Parallel.ForEach(1, CHighPrime)
    .Initialize(
      procedure (var taskState: TOmniValue)
      begin
        taskState.AsInteger := 0;
      end)
    .Finalize(
```

```
13        procedure (const taskState: TOmniValue)
14        begin
15          lockNum.Acquire;
16          try
17            numPrimes := numPrimes + taskState.AsInteger;
18          finally lockNum.Release; end;
19        end)
20      .Execute(
21        procedure (const value: integer; var taskState: TOmniValue)
22        begin
23          if IsPrime(value) then
24            taskState.AsInteger := taskState.AsInteger + 1;
25        end
26      );
27  end;
```

2.3.9 Handling Exceptions

`ForEach` abstraction does not yet implement any exception handling. You should always wrap task method (code passed to the `Execute`) in `try..except` if you expect the code to raise exceptions.

2.3.10 Examples

Practical example of *For Each* usage can be found in chapters Parallel For with Synchronized Output and Parallel Search in a Tree.

3. Low-level Multithreading

The low-level OmniThreadLibrary layer focuses on the `task` concept. In most aspects this is similar to the Delphi's TThread approach except that OmniThreadLibrary focuses on the code (a.k.a. task) and interaction with the code while the Delphi focuses on the operating system primitive required for executing additional threads (TThread).

A task is created using the `CreateTask` function, which takes as a parameter a global procedure, a method, an instance of the `TOmniWorker` class (or, usually, a descendant of that class) or an anonymous procedure (in Delphi 2009 and newer). `CreateTask` will also accept an optional second parameter, a task name, which will be displayed in the Delphi's Thread view on the thread running the task.

```
1  type
2    TOmniTaskProcedure = procedure(const task: IOmniTask);
3    TOmniTaskMethod = procedure(const task: IOmniTask) of object;
4    TOmniTaskDelegate = reference to procedure(const task: IOmniTask);
5
6  function CreateTask(worker: TOmniTaskProcedure; const taskName: string = ''):
7    IOmniTaskControl; overload;
8  function CreateTask(worker: TOmniTaskMethod; const taskName: string = ''):
9    IOmniTaskControl; overload;
10 function CreateTask(worker: TOmniTaskDelegate; const taskName: string = ''):
11   IOmniTaskControl; overload;
12 function CreateTask(const worker: IOmniWorker; const taskName: string = ''):
13   IOmniTaskControl; overload;
```

`CreateTask` returns a feature-full interface `IOmniTaskControl` which we will explore in this chapter. The most important function in this interface, `Run`, will create a new thread and start your task in it.

3.1 Low-level For the Impatient

The following code represents the simplest possible low-level OmniThreadLibrary example. It executes the `Beep` function in a background thread. The `Beep` function merely beeps and exits. By exiting from the task function, the Windows thread running the task is also terminated.

```
procedure TfrmTestSimple.Beep(const task: IOmniTask);
begin
  //Executed in a background thread
  MessageBeep(MB_ICONEXCLAMATION);
end;

CreateTask(Beep, 'Beep').Run;
```

Another way to start a task is to call a `Schedule` function which starts it in a thread allocated from a thread pool. This is covered in the Thread Pooling chapter.

3.2 Four ways to create a task

Let's examine all four ways of creating a task. The simplest possible way (demoed in application `2_TwoWayHello`) is to pass a name of a global procedure to the `CreateTask`. This global procedure must consume one parameter of type `IOmniTask` .

```
procedure RunHelloWorld(const task: IOmniTask);
begin
  //
end;

CreateTask(RunHelloWorld, 'HelloWorld').Run;
```

A variation on the theme is passing a name of a method to the CreateTask. This approach is used in the demo application `1_HelloWorld`. The interesting point here is that you can declare this method in the same class from which the `CreateTask` is called. That way you can access all class fields and methods from the threaded code. Just keep in mind that you'll be doing this from another thread so make sure you know what you're doing!

```
procedure TfrmTestHelloWorld.RunHelloWorld(const task: IOmniTask);
begin
  //
end;

procedure TfrmTestHelloWorld.StartTask;
begin
  CreateTask(RunHelloWorld, 'HelloWorld').Run;
end;
```

In Delphi 2009 and newer you can also write the task code as an anonymous function.

```
CreateTask(
  procedure (const task: IOmniTask)
  begin
    //
  end,
  'HellowWorld').Run;
```

For all except the simplest tasks, you'll use the fourth approach as it will give you access to the true OmniThreadLibrary power (namely internal wait loop and message dispatching). To use it, you have to create a worker object deriving from the `TOmniWorker` class.

```
type
  THelloWorker = class(TOmniWorker)
  end;

procedure TfrmTestTwoWayHello.actStartHelloExecute(Sender: TObject);
begin
  FHelloTask :=
    CreateTask(THelloWorker.Create(), 'Hello').
    Run;
end;
```

3.3 IOmniTaskControl and IOmniTask Interfaces

When you create a low-level task, OmniThreadLibrary returns a *task controller* interface `IOmniTaskControl`. This interface, which is defined in the *OtlTaskControl* unit, can be used to control the task from the owner's side. The task code, on the other hand, has access to another interface, `IOmniTask` (defined in the *OtlTask* unit), which can be used to communicate with the owner and manipulate the task itself. A picture in the Tasks vs. Threads chapter shows the relationship between those interfaces.

This chapter deals mainly with these two interfaces. For the reference reasons, the `IOmniTaskControl` is reprinted here in full. In the rest of the chapter I'll just show relevant interface parts.

The `IOmniTask` interface is described at the end of this chapter.

```
type
  IOmniTaskControl = interface
    function  Alertable: IOmniTaskControl;
    function  CancelWith(const token: IOmniCancellationToken): IOmniTaskControl;
    function  ChainTo(const task: IOmniTaskControl;
      ignoreErrors: boolean = false): IOmniTaskControl;
    function  ClearTimer(timerID: integer): IOmniTaskControl;
    function  DetachException: Exception;
    function  Enforced(forceExecution: boolean = true): IOmniTaskControl;
    function  GetFatalException: Exception;
```

```
11      function  GetParam: TOmniValueContainer;
12      function  Invoke(const msgMethod: pointer): IOmniTaskControl; overload;
13      function  Invoke(const msgMethod: pointer;
14        msgData: array of const): IOmniTaskControl; overload;
15      function  Invoke(const msgMethod: pointer;
16        msgData: TOmniValue): IOmniTaskControl; overload;
17      function  Invoke(const msgName: string): IOmniTaskControl; overload;
18      function  Invoke(const msgName: string;
19        msgData: array of const): IOmniTaskControl; overload;
20      function  Invoke(const msgName: string;
21        msgData: TOmniValue): IOmniTaskControl; overload;
22      function  Invoke(remoteFunc: TOmniTaskControlInvokeFunction):
23        IOmniTaskControl; overload;
24      function  Invoke(remoteFunc: TOmniTaskControlInvokeFunctionEx):
25        IOmniTaskControl; overload;
26      function  Join(const group: IOmniTaskGroup): IOmniTaskControl;
27      function  Leave(const group: IOmniTaskGroup): IOmniTaskControl;
28      function  MonitorWith(const monitor: IOmniTaskControlMonitor):
29        IOmniTaskControl;
30      function  MsgWait(wakeMask: DWORD = QS_ALLEVENTS): IOmniTaskControl;
31      function  OnMessage(eventDispatcher: TObject): IOmniTaskControl; overload;
32      function  OnMessage(eventHandler: TOmniTaskMessageEvent): IOmniTaskControl; overload;
33      function  OnMessage(msgID: word; eventHandler: TOmniTaskMessageEvent):
34        IOmniTaskControl; overload;
35      function  OnMessage(msgID: word; eventHandler: TOmniMessageExec):
36        IOmniTaskControl; overload;
37      function  OnMessage(eventHandler: TOmniOnMessageFunction):
38        IOmniTaskControl; overload;
39      function  OnMessage(msgID: word; eventHandler: TOmniOnMessageFunction):
40        IOmniTaskControl; overload;
41      function  OnTerminated(eventHandler: TOmniOnTerminatedFunction):
42        IOmniTaskControl; overload;
43      function  OnTerminated(eventHandler: TOmniOnTerminatedFunctionSimple):
44        IOmniTaskControl; overload;
45      function  OnTerminated(eventHandler: TOmniTaskTerminatedEvent):
46        IOmniTaskControl; overload;
47      function  RemoveMonitor: IOmniTaskControl;
48      function  Run: IOmniTaskControl;
49      function  Schedule(const threadPool: IOmniThreadPool = nil {default pool}):
50        IOmniTaskControl;
51      function  SetMonitor(hWindow: THandle): IOmniTaskControl;
52      function  SetParameter(const paramName: string;
53        const paramValue: TOmniValue): IOmniTaskControl; overload;
54      function  SetParameter(const paramValue: TOmniValue):
55        IOmniTaskControl; overload;
```

```
56      function  SetParameters(const parameters: array of TOmniValue):
57        IOmniTaskControl;
58      function  SetPriority(threadPriority: TOTLThreadPriority): IOmniTaskControl;
59      function  SetQueueSize(numMessages: integer): IOmniTaskControl;
60      function  SetTimer(timerID: integer; interval_ms: cardinal;
61        const timerMessage: TOmniMessageID): IOmniTaskControl; overload;
62      function  SetUserData(const idxData: TOmniValue;
63        const value: TOmniValue): IOmniTaskControl;
64      procedure Stop;
65      function  Terminate(maxWait_ms: cardinal = INFINITE): boolean;
66      function  TerminateWhen(event: THandle): IOmniTaskControl; overload;
67      function  TerminateWhen(token: IOmniCancellationToken):
68        IOmniTaskControl; overload;
69      function  Unobserved: IOmniTaskControl;
70      function  WaitFor(maxWait_ms: cardinal): boolean;
71      function  WaitForInit: boolean;
72      function  WithCounter(const counter: IOmniCounter): IOmniTaskControl;
73      function  WithLock(const lock: TSynchroObject;
74        autoDestroyLock: boolean = true): IOmniTaskControl; overload;
75      function  WithLock(const lock: IOmniCriticalSection):
76        IOmniTaskControl; overload;
77    //
78      property CancellationToken: IOmniCancellationToken
79        read GetCancellationToken;
80      property Comm: IOmniCommunicationEndpoint read GetComm;
81      property ExitCode: integer read GetExitCode;
82      property ExitMessage: string read GetExitMessage;
83      property FatalException: Exception read GetFatalException;
84      property Lock: TSynchroObject read GetLock;
85      property Name: string read GetName;
86      property Param: TOmniValueContainer read GetParam;
87      property UniqueID: int64 read GetUniqueID;
88      property UserData[const idxData: TOmniValue]: TOmniValue
89        read GetUserDataVal write SetUserDataVal;
90    end;
```

3.4 Task Controller Needs an Owner

The `IOmniTaskController` interface returned from the `CreateTask` must always be stored in a variable/field with a scope that exceeds the lifetime of the background task. In other words, don't store a long-term background task interface in a local variable.

The simplest example of the wrong approach can be written in one line:

```
CreateTask(MyWorker).Run;
```

This code looks fine, but it doesn't work. In this case, the `IOmniTaskController` interface is stored in a hidden temporary variable which is destroyed at the end of the current method. This then causes the task controller to be destroyed which in turn causes the background task to be destroyed. Running this code would therefore just create and then destroy the task.

A common solution is to just store the interface in some field.

```
FTaskControl := CreateTask(MyWorker).Run;
```

When you don't need background worker anymore, you should terminate the task and free the task controller.

```
FTaskControl.Terminate;
FTaskControl := nil;
```

Another solution is to provide the task with an implicit owner. You can, for example, use the event monitor to monitor tasks lifetime or messages sent from the task and that will make the task owned by the monitor. The following code is therefore valid:

```
CreateTask(MyWorker).MonitorWith(eventMonitor).Run;
```

Yet another possibility is to call the `Unobserved`) before the `Run`. This method makes the task being observed by an internal monitor.

```
CreateTask(MyWorker).Unobserved.Run;
```

When you use a thread pool to run a task, the thread pool acts as a task owner so there's no need for an additional explicit owner.

```
procedure Beep(const task: IOmniTask);
begin
  MessageBeep(MB_ICONEXCLAMATION);
end;

CreateTask(Beep, 'Beep').Schedule;
```

3.5 Communication Subsystem

As it is explained in the Locking vs. Messaging section, OmniThreadLibrary automatically creates a communication channel between the task controller and the task and exposes it through the `Comm` property. The communication channel is not exclusive to the OmniThreadLibrary; you could use it equally well from a `TThread`-based multithreading code.

```
property Comm: IOmniCommunicationEndpoint read GetComm;
```

The `IOmniCommunicationEndpoint` interface exposes a simple interface for sending and receiving messages.[1]

`IOmniCommunicationEndpoint` are described at http://www.thedelphigeek.com/2008/07/omnithreadlibrary-internals-otlcomm.html.

```
type
  TOmniMessage = record
    MsgID  : word;
    MsgData: TOmniValue;
    constructor Create(aMsgID: word; aMsgData: TOmniValue); overload;
    constructor Create(aMsgID: word); overload;
  end;

  IOmniCommunicationEndpoint = interface
    function  Receive(var msg: TOmniMessage): boolean; overload;
    function  Receive(var msgID: word; var msgData: TOmniValue): boolean; overload;
    function  ReceiveWait(var msg: TOmniMessage; timeout_ms: cardinal): boolean; overload;
    function  ReceiveWait(var msgID: word; var msgData: TOmniValue;
      timeout_ms: cardinal): boolean; overload;
    procedure Send(const msg: TOmniMessage); overload;
    procedure Send(msgID: word); overload;
    procedure Send(msgID: word; msgData: array of const); overload;
    procedure Send(msgID: word; msgData: TOmniValue); overload;
    function  SendWait(msgID: word;
      timeout_ms: cardinal = CMaxSendWaitTime_ms): boolean; overload;
    function  SendWait(msgID: word; msgData: TOmniValue;
      timeout_ms: cardinal = CMaxSendWaitTime_ms): boolean; overload;
    property NewMessageEvent: THandle read GetNewMessageEvent;
    property OtherEndpoint: IOmniCommunicationEndpoint read GetOtherEndpoint;
    property Reader: TOmniMessageQueue read GetReader;
    property Writer: TOmniMessageQueue read GetWriter;
  end;
```

- `Receive`

  Both variants of `Receive` return the first message from the message queue, either as a `TOmniMessage` record or as a (*message ID*, *message data*) pair. Data is always passed as a `TOmniValue` record.

  The function returns a `True` if a message was returned, `False` if the message queue is empty.

- `ReceiveWait`

  These two variations of the `Receive` allow you to specify the maximum timeout (in milliseconds) you are willing to wait for the next message. Timeout of `0` milliseconds makes the function behave just like

[1] Internal workings of the

the `Receive`. Special timeout value `INFINITE` (defined in `Windows.pas`) will make the function wait as long as necessary.

The function returns `True` if a message was returned, `False` if the message queue is still empty after the timeout.

- `Send`

  Four overloaded versions of `Send` all write a message to the message queue and raise an exception if the queue is full. [Message queue size defaults to 1000 elements and can be increased by calling the `OmniTaskControl.SetQueueSize` before the communication channel is used for the first time.]

  The `Send(msgID: word)` version sends an empty message data (`TOmniValue.Null`).

  The `Send(msgID: word; msgData: array of const)` version packs the data array into one `TOmniValue` value by calling `TOmniValue.Create(msgData)`.

- `SendWait`

  These two variations of the `Send` allow you to specify the maximum timeout (in milliseconds) you are willing to wait if a message queue is full and there's no place for the messages. The timeout of `0` ms makes the function behave just like the `Send`. A timeout of `INFINITE` milliseconds is also supported.

  The function returns `True` if message was successfully sent, `False` if the message queue is still full after the timeout.

- `NewMessageEvent`

  This property returns Windows *event* which is signalled every time new data is inserted in the queue. This event is not created until the code accesses the `NewMessageEvent` property for the first time.

- `OtherEndpoint`

  Returns the other end of the communication channel (task's end if accessed through the `IOmniTaskControl.Comm` and task controller's end if accessed through the `IOmniTask.Comm` interface).

- `Reader`

  Returns the input queue associated with this endpoint.

- `Writer`

  Returns the output queue associated with this endpoint.

In versions up to [3.04a], both `SendWait` and `ReceiveWait` were designed to be used from only one thread at a time. Since OmniThreadLibrary [3.04b] they are both fully thread-safe and can be used from multiple producers and consumers at the same time.

For a practical examples on communication channel usage, see the Communication subsection of simple tasks and `TOmniWorker` tasks sections.

Communication message queue is implemented using the Bounded Queue structure.

3.6 Lock-free Collections

OmniThreadLibrary implements three lock-free data structures which are very suitable for low-level usage – bounded stack, bounded queue and dynamic queue. Bounded queue is used inside the OmniThreadLibrary for messaging and dynamic queue is used as a basis of the blocking collection. All three are implemented in the *OtlContainers* unit.

Another lock-free data structure, a message queue, is defined in the *OtlComm* unit and is mostly intended for internal operation (such as sending messages to and from thread) although it can also be used for other tasks. An example of such usage is shown in the Using Message Queue with a TThread Worker chapter.

> The term *lock-free* is not well defined (and not even universally accepted). In the context of this book lock-free means that the synchronisation between threads is not achieved with the user- or kernel-level synchronisation primitives such as critical sections, but with bus-locking CPU instructions. With modern CPU architectures this approach is much faster than locking on the operating system level.

All three data structures are fully thread-safe. They support multiple simultaneous readers and writers.

See also demos `10_Containers` *and* `32_Queue`.

3.6.1 Bounded Stack

The bounded stack[2] structure is a very fast stack with limited length. The core of the implementation is stored in the `TOmniBaseBoundedStack` class.

Derived class `TOmniBoundedStack` adds support for external observers. Both classes implement the same interface – `IOmniStack` – so you can code against the class or against the interface.

```
1  type
2    IOmniStack = interface
3      procedure Empty;
4      procedure Initialize(numElements, elementSize: integer);
5      function  IsEmpty: boolean;
6      function  IsFull: boolean;
7      function  Pop(var value): boolean;
8      function  Push(const value): boolean;
9    end;
10
11   TOmniBaseBoundedStack = class(TInterfacedObject, IOmniStack)
12   ...
13   public
14     destructor Destroy; override;
15     procedure Empty;
```

[2] http://en.wikipedia.org/wiki/Stack_%28abstract_data_type%29

```
16      procedure Initialize(numElements, elementSize: integer); virtual;
17      function  IsEmpty: boolean; inline;
18      function  IsFull: boolean; inline;
19      function  Pop(var value): boolean;
20      function  Push(const value): boolean;
21      property  ElementSize: integer read obsElementSize;
22      property  NumElements: integer read obsNumElements;
23    end;
24
25    TOmniBoundedStack = class(TOmniBaseBoundedStack)
26    ...
27    public
28      constructor Create(numElements, elementSize: integer;
29        partlyEmptyLoadFactor: real = CPartlyEmptyLoadFactor;
30        almostFullLoadFactor: real = CAlmostFullLoadFactor);
31      destructor  Destroy; override;
32      function Pop(var value): boolean;
33      function Push(const value): boolean;
34      property ContainerSubject: TOmniContainerSubject read osContainerSubject;
35    end;
```

- `Empty`

  Empties the stack.
- `Initialize`

  Initializes the stack for maximum `numElements` elements of size `elementSize`.
- `IsEmpty`

  Returns `True` when the stack is empty.
- `IsFull`

  Returns `True` when the stack is full.
- `Pop`

  Takes one value from the stack and returns `True` if the stack was not empty before the operation.
- `Push`

  Puts one value on the stack and returns `True` if there was place for the value (the stack was not full before the operation).
- `ElementSize`

  Returns the size of the stack element as set in the `Initialize` call.
- `NumElements`

  Returns maximum number of elements in the stack as set in the `Initialize` call.
- `ContainerSubject`

  Provides a point for attaching external observers as described in the Observing Lock-free Collections section.

3.6.2 Bounded Queue

The bounded queue[3] structure is a very fast queue with limited length.

The core of the implementation is stored in the `TOmniBaseBoundedQueue` class. Derived class `TOmniBoundedQueue` adds support for external observers. Both classes implement the same interface – `IOmniQueue` – so you can code against the class or against the interface.

```
type
  IOmniQueue = interface
    function  Dequeue(var value): boolean;
    procedure Empty;
    function  Enqueue(const value): boolean;
    procedure Initialize(numElements, elementSize: integer);
    function  IsEmpty: boolean;
    function  IsFull: boolean;
  end;

  TOmniBaseBoundedQueue = class(TInterfacedObject, IOmniQueue)
  ...
  public
    destructor Destroy; override;
    function  Dequeue(var value): boolean;
    procedure Empty;
    function  Enqueue(const value): boolean;
    procedure Initialize(numElements, elementSize: integer); virtual;
    function  IsEmpty: boolean;
    function  IsFull: boolean;
    property  ElementSize: integer read obqElementSize;
    property  NumElements: integer read obqNumElements;
  end;

  TOmniBoundedQueue = class(TOmniBaseBoundedQueue)
  ...
  public
    constructor Create(numElements, elementSize: integer;
      partlyEmptyLoadFactor: real = CPartlyEmptyLoadFactor;
      almostFullLoadFactor: real = CAlmostFullLoadFactor);
    destructor  Destroy; override;
    function  Dequeue(var value): boolean;
    function  Enqueue(const value): boolean;
    property  ContainerSubject: TOmniContainerSubject read oqContainerSubject;
  end;
```

- Empty

[3] http://en.wikipedia.org/wiki/Queue_%28data_structure%29

Empties the stack.

- `Dequeue`

  Takes one value from the queue's head and returns `True` if the queue was not empty before the operation.

- `Enqueue`

  Inserts one value on the queue's tail and returns `True` if there was place for the value (the queue was not full before the operation).

- `Initialize`

  Initializes the queue for maximum `numElements` elements of size `elementSize`.

- `IsEmpty`

  Returns `True` when the queue is empty.

- `IsFull`

  Returns `True` when the queue is full.

- `ElementSize`

  Returns size of the queue element as set in the `Initialize` call.

- `NumElements`

  Returns maximum number of elements in the queue as set in the `Initialize` call.

- `ContainerSubject`

  Provides a point for attaching external observers as described in the Observing Lock-free Collections section.

3.6.3 Message Queue

The `TOmniMessageQueue` is just a thin wrapper around the bounded queue data structure. An element of this queue is a (*message ID*, *message data*) pair, stored in a `TOmniMessage` record.

This class greatly simplifies creating and attaching event and window observers.

```
type
  TOmniMessage = record
    MsgID  : word;
    MsgData: TOmniValue;
    constructor Create(aMsgID: word; aMsgData: TOmniValue); overload;
    constructor Create(aMsgID: word); overload;
  end;

  TOmniContainerWindowsEventObserver = class(TOmniContainerObserver)
  public
    function  GetEvent: THandle; virtual; abstract;
  end;

  TOmniMessageQueueMessageEvent =
```

```
15      procedure(Sender: TObject; const msg: TOmniMessage) of object;
16
17    TOmniMessageQueue = class(TOmniBoundedQueue)
18    public
19      constructor Create(numMessages: integer;
20        createEventObserver: boolean = true); reintroduce;
21      destructor  Destroy; override;
22      function  Dequeue: TOmniMessage; reintroduce;
23      function  Enqueue(const value: TOmniMessage): boolean; reintroduce;
24      procedure Empty;
25      function  GetNewMessageEvent: THandle;
26      function  TryDequeue(var msg: TOmniMessage): boolean; reintroduce;
27      property EventObserver: TOmniContainerWindowsEventObserver
28        read mqWinEventObserver;
29      property OnMessage: TOmniMessageQueueMessageEvent
30        read mqWinMsgObserver.OnMessage write SetOnMessage;
31    end;
```

`TOmniMessageQueue.Create` creates an event observer unless the second parameter (`createEventObserver`) is set to `False`. It is created with the `coiNotifyOnAllInserts` interest meaning that an event (accessible through the `GetNewMessageEvent` function) is signalled each time an element (a message) is added to the queue. The observer itself is accessible through the `EventObserver` property.

You can also easily create a window message observer by attaching an event handler to the `OnMessage` property. This observer is also created with the `coiNotifyOnAllInserts` interest which causes the `OnMessage` event handler to be called each time an element (a message) is added to the queue. You can destroy this observer at any time by assigning a `nil` value to the `OnMessage` event.

For an example, see chapter Using Message Queue with a TThread Worker.

3.6.4 Dynamic Queue

The dynamic queue[4] is a fast queue with unlimited length. It can grow as much as needed as the data used to store elements is dynamically allocated.

The core of the implementation is stored in the `TOmniBaseQueue` class. Derived class `TOmniQueue` adds support for external observers. Both structures store TOmniValue elements.

[4] http://en.wikipedia.org/wiki/Queue_%28data_structure%29

```
type
  TOmniBaseQueue = class
  ...
  public
    constructor Create(blockSize: integer = 65536; numCachedBlocks: integer = 4);
    destructor  Destroy; override;
    function  Dequeue: TOmniValue;
    procedure Enqueue(const value: TOmniValue);
    function  IsEmpty: boolean;
    function  TryDequeue(var value: TOmniValue): boolean;
  end;

  TOmniQueue = class(TOmniBaseQueue)
  ...
  public
    function  Dequeue: TOmniValue;
    procedure Enqueue(const value: TOmniValue);
    function  TryDequeue(var value: TOmniValue): boolean;
    property ContainerSubject: TOmniContainerSubject read ocContainerSubject;
  end;
```

- `Create`

  Creates a queue object with a specified page size (`blockSize`) where `numCachedBlocks` are always preserved for future use. Defaults (`65536` and `4`) should be appropriate for most scenarios.
- `Dequeue`

  Takes one element from queue's head and returns it. If the queue is empty, an exception is raised.
- `Enqueue`

  Inserts an element on the queue's tail.
- `IsEmpty`

  Returns `True` when the queue is empty.
- `TryDequeue`

  Takes one element from queue's head and returns it in the `value` parameter. Returns `True` if an element was returned (the queue was not empty before the operation).
- `ContainerSubject`

  Provides a point for attaching external observers as described in the Observing Lock-free Collections section.

3.6.5 Observing Lock-free Collections

OmniThreadLibrary data structures support the *observer*[5] design pattern. Each structure can be observed by multiple observers at the same time. Supporting code and two practical observers are stored in the *OtlContainerObserver* unit.

[5] http://en.wikipedia.org/wiki/Observer_pattern

Current architecture supports four different kinds of events that can be observed:

```
type
  ///<summary>All possible actions observer can take interest in.</summary>
  TOmniContainerObserverInterest = (
    //Interests with permanent subscription:
    coiNotifyOnAllInserts, coiNotifyOnAllRemoves,
    //Interests with one-shot subscription:
    coiNotifyOnPartlyEmpty, coiNotifyOnAlmostFull
  );
```

- `coiNotifyOnAllInserts`

  Observer is notified whenever a data element is inserted into the structure.
- `coiNotifyOnAllRemoves`

  Observer is notified whenever a data element is removed from the structure.
- `coiNotifyOnPartlyEmpty`

  Observer is notified whenever a data usage drops below the `partlyEmptyLoadFactor` (parameter of the data structure constructor, 80% by default). This event is only supported for bounded structures.

  This event can only be observed once. After that you should destroy the observer and (if required) create another one and attach it to the data structure.
- `coiNotifyOnAlmostFull`

  Observer is notified whenever a data usage rises above the `almostFullLoadFactor` (parameter of the data structure constructor, 90% by default). This event is only supported for bounded structures.

  This event can only be observed once. After that you should destroy the observer and (if required) create another one and attach it to the data structure.

The *OtlContainerObserver* unit implements two kinds of observers.

```
  TOmniContainerWindowsEventObserver = class(TOmniContainerObserver)
  public
    function  GetEvent: THandle; virtual; abstract;
  end;

  TOmniContainerWindowsMessageObserver = class(TOmniContainerObserver)
  strict protected
    function  GetHandle: THandle; virtual; abstract;
  public
    procedure Send(aMessage: cardinal; wParam, lParam: integer);
      virtual; abstract;
    property Handle: THandle read GetHandle;
  end;

```

```
15    function CreateContainerWindowsEventObserver(externalEvent: THandle = 0):
16      TOmniContainerWindowsEventObserver;
17
18    function CreateContainerWindowsMessageObserver(hWindow: THandle;
19      msg: cardinal; wParam, lParam: integer):
20      TOmniContainerWindowsMessageObserver;
```

The *event* observer `TOmniContainerWindowsEventObserver` raises an event every time the observed event occurs.

The *window message* observer `TOmniContainerWindowMessageObserver` sends a message to a window every time the observed event occurs.

3.6.5.1 Examples

Create and attach the *event* observer:

```
1  FObserver := CreateContainerWindowsEventObserver;
2  FCollection.ContainerSubject.Attach(FObserver, coiNotifyOnAllInserts);
```

Access the observer event so you can wait on it:

```
1  FObserver.GetEvent;
```

Detach and destroy the observer:

```
1  FCollection.ContainerSubject.Detach(FObserver, coiNotifyOnAllInserts);
2  FreeAndNil(FObserver);
```

Create and attach the *window message* observer:

```
1  FWindow := DSiAllocateHWnd(ObserverWndProc);
2  FObserver := CreateContainerWindowsMessageObserver(
3    FWindow, MSG_ITEM_INSERTED, 0, 0);
4  FWorker.Output.ContainerSubject.Attach(FObserver, coiNotifyOnAllInserts);
```

Process observer messages:

```
procedure ObserverWndProc(var message: TMessage);
var
  ovWorkItem: TOmniValue;
  workItem  : IOmniWorkItem;
begin
  if message.Msg = MSG_ITEM_INSERTED then begin
    //...
    message.Result := Ord(true);
  end
  else
    message.Result := DefWindowProc(FWindow, message.Msg,
      message.WParam, message.LParam);
end;
```

Detach and destroy the observer:

```
FWorker.Output.ContainerSubject.Detach(FObserver, coiNotifyOnAllInserts);
FreeAndNil(FObserver);
DSiDeallocateHWnd(FWindow);
```

3.6.6 Benchmarks

OmniThreadLibrary contains two demos that can be used to measure the performance of the lock-free structures. Bounded structures are benchmarked in the `10_Containers` demo and dynamic queue is benchmarked in the `32_Queue` demo.

Following results were measured on 4-core i7-2630QM running at 2 GHz. As you can see, lock-free structures can transfer from 2,5 to 5 million messages per second.


OtlContainers tester
Base stack stress test (1 -> 1)
2 -> 1
1 -> 2
1 -> 4
4 -> 1
2 -> 2
4 -> 4
Base s. correctness test
Run all
Stack stress test (1 -> 1)
2 -> 1
1 -> 2
1 -> 4
4 -> 1
2 -> 2
4 -> 4
Stack correctness test
Run all
Base queue stress test (1 -> 1)
2 -> 1
1 -> 2
1 -> 4
4 -> 1
2 -> 2
4 -> 4
Base q. correctness test
Run all
Queue stress test (1 -> 1)
2 -> 1
1 -> 2
1 - > 4
4 -> 1
2 -> 2
4 -> 4
Queue correctness test
Run all
Run all tests
Test duration (sec):
60
Save log
Clear log
10:46 32-bit
10:47 Starting 60 second base stack stress test, 2 -> 2
10:48 Writer completed in 60033 ms; 72568833 pushed, 17739774 skipped; 1208816 msg/s
10:48 Reader completed in 60035 ms; 74299064 popped, 3557703 empty; 1237596 msg/s
10:48 Reader completed in 60036 ms; 72946222 popped, 3468753 empty; 1215041 msg/s
10:48 Writer completed in 60037 ms; 74677453 pushed, 18121522 skipped; 1243857 msg/s
10:48 All tasks stopped
10:48 Total reader throughput: 2452637 msg/s
10:48 Total writer throughput: 2452673 msg/s
10:48 Writing to queue
10:48 Reading from queue
10:48 Test completed
10:48 Starting 60 second stack stress test, 2 -> 2
10:49 Writer completed in 60011 ms; 72685244 pushed, 1239363 skipped; 1211199 msg/s
10:49 Writer completed in 60014 ms; 73432535 pushed, 1278504 skipped; 1223590 msg/s
10:49 Reader completed in 60021 ms; 73102234 popped, 429157 empty; 1217944 msg/s
10:49 Reader completed in 60022 ms; 73015545 popped, 450310 empty; 1216480 msg/s
10:49 All tasks stopped
10:49 Total reader throughput: 2434424 msg/s
10:49 Total writer throughput: 2434789 msg/s
10:49 Writing to queue
10:49 Reading from queue
10:49 Test completed
10:50 Starting 60 second base queue stress test, 2 -> 2
10:51 Writer completed in 60003 ms; 131932744 enqueued, 1039799 skipped; 2198769 msg/s
10:51 Reader completed in 60007 ms; 132167857 dequeued, 18696014 empty; 2202541 msg/s
10:51 Writer completed in 60007 ms; 132447228 enqueued, 1049603 skipped; 2207196 msg/s
10:51 Reader completed in 60015 ms; 132212115 dequeued, 18717292 empty; 2202985 msg/s
10:51 All tasks stopped
10:51 Total reader throughput: 4405526 msg/s
10:51 Total writer throughput: 4405965 msg/s
10:51 Writing to queue
10:51 Reading from queue
10:51 Test completed
10:51 Starting 60 second queue stress test, 2 -> 2
10:52 Writer completed in 60018 ms; 114878306 enqueued, 6301 skipped; 1914064 msg/s
10:52 Reader completed in 60027 ms; 114876010 dequeued, 8597 empty; 1913739 msg/s
10:52 Reader completed in 60028 ms; 114941534 dequeued, 8609 empty; 1914799 msg/s
10:52 Writer completed in 60029 ms; 114940238 enqueued, 9905 skipped; 1914745 msg/s
10:52 All tasks stopped
10:52 Total reader throughput: 3828538 msg/s
10:52 Total writer throughput: 3828809 msg/s
10:52 Writing to queue
10:52 Reading from queue
10:52 Test completed


TOmniQueue tester
Single-threaded test
1 -> 1
2 -> 2
3 -> 3
4 -> 4
8 -> 8
1 -> 7
7 -> 1
Repeat
Queue class
TOmniBaseQueue
TOmniQueue
Test IsEmpty
Block size:
64 KB
Send interfaces
[11:03:34] 2 -> 2
[11:03:37] All worker threads terminated, execution time = 1965, 5089059 msgs/s
[11:03:52] 2 -> 2
[11:03:55] All worker threads terminated, execution time = 2009, 4977601 msgs/s
[11:04:00] 2 -> 2
[11:04:03] All worker threads terminated, execution time = 1981, 5047956 msgs/s

4. Synchronization

Although the OmniThreadLibrary treats communication as a superior approach to locking, there are still times when using "standard" synchronization primitives such as a critical section are unavoidable. As the standard Delphi/Windows approach to locking is very low-level, OmniThreadLibrary builds on it and improves it in some significant ways. All these improvements are collected in the *OtlSync* unit and are described in the following sections. The only exception is the waitable value class/interface, which is declared in the *OtlCommon* unit.

This part of the book assumes that you have a basic understanding of locking. If you are new to the topic, you should first read the appropriate chapters from one of the books mentioned in the introduction.

4.1 Critical Sections

The most useful synchronisation primitive for multithreaded programming is indubitably the *critical section*[1].

OmniThreadLibrary simplifies sharing critical sections between a task owner and a task with the use of the `WithLock` method. High-level tasks can access this method through the task configuration block.

I was always holding the opinion that locks should be as granular as possible. Putting many small locks around many unrelated pieces of code is better than using one giant lock for everything, but programmers frequently use one or few locks because managing many critical sections can be a bother.

To help you with writing a better code, OmniThreadLibrary implements three extensions to the Delphi's TCriticalSection class - `IOmniCriticalSection`, `TOmniCS` and `Locked<T>`.

4.1.1 IOmniCriticalSection

Delphi implements critical section support with a TCriticalSection class which must be created and destroyed in the code. (There is also a TRTLCriticalSection record, but it is only supported on Windows.) OmniThreadLibrary extends this implementation with an `IOmniCriticalSection` interface, which you only have to create. The compiler will make sure that it is destroyed automatically at the appropriate place.

[1] http://en.wikipedia.org/wiki/Critical_section, http://docwiki.embarcadero.com/Libraries/XE3/en/System.SyncObjs.TCriticalSection

```
type
  IOmniCriticalSection = interface
    procedure Acquire;
    procedure Release;
    function  GetSyncObj: TSynchroObject;
    property LockCount: integer read GetLockCount;
  end;

function CreateOmniCriticalSection: IOmniCriticalSection;
```

`IOmniCriticalSection` uses `TCriticalSection` internally. It acts just as a proxy that calls `TCriticalSection` functions. Besides that, it provides an additional functionality by counting the number of times a critical section has been acquired, which can help a lot while debugging. This counter can be read through the `LockCount` property.

> A critical section can be *acquired* multiple times from one thread. For example, the following code is perfectly valid:

```
cSec := CreateOmniCriticalSection; //LockCount = 0
cSec.Acquire; //LockCount = 1
cSec.Acquire; //LockCount = 2
cSec.Release; //LockCount = 1
cSec.Release; //LockCount = 0
```

Additionally, `IOmniCriticalSection` doesn't use `TCriticalSection` directly, but wraps it into a larger object as suggested by Eric Grange[2].

4.1.2 TOmniCS

Another `TCriticalSection` extension found in the OmniThreadLibrary is the `TOmniCS` record. It allows you to use a critical section by simply declaring a record in appropriate place.

Using `TOmniCS`, locking can be as simple as this:

```
uses
  GpLists,
  OtlSync;

procedure ProcessList(const intf: IGpIntegerList);
begin
  //...
end;

```

[2] http://delphitools.info/2011/11/30/fixing-tcriticalsection/

```
var
  lock: TOmniCS;
  intf: IGpIntegerList;

procedure Test1;
begin
  intf := TGpIntegerList.Create;
  //...
  lock.Acquire;
  try
    ProcessList(intf);
  finally lock.Release; end;
end;
```

`TOmniCS` is implemented as a record with one private field holding the `IOmniCriticalSection` interface.

```
type
  TOmniCS = record
  strict private
    ocsSync: IOmniCriticalSection;
  private
    function  GetLockCount: integer; inline;
    function  GetSyncObj: TSynchroObject; inline;
  public
    procedure Initialize;
    procedure Acquire; inline;
    procedure Release; inline;
    property LockCount: integer read GetLockCount;
    property SyncObj: TSynchroObject read GetSyncObj;
  end;
```

The `Release` method merely calls the `Release` method on the internal interface, while the `Acquire` method is more tricky as it has to initialize the `ocsSync` field first.

```
procedure TOmniCS.Acquire;
begin
  Initialize;
  ocsSync.Acquire;
end;

procedure TOmniCS.Release;
begin
  ocsSync.Release;
end;
```

The initialization, hidden inside the `Initialize` method (which you can also call from the code to initialize the critical section explicitly) is quite tricky because it has to initialize the `ocsSync` only once and must work properly when called from two places (two threads) at the same time. This is achieved by using the *optimistic initialization* approach, described later in this chapter.

```
procedure TOmniCS.Initialize;
var
  syncIntf: IOmniCriticalSection;
begin
  Assert(cardinal(@ocsSync) mod SizeOf(pointer) = 0,
    'TOmniCS.Initialize: ocsSync is not properly aligned!');
  Assert(cardinal(@syncIntf) mod SizeOf(pointer) = 0,
    'TOmniCS.Initialize: syncIntf is not properly aligned!');
  if not assigned(ocsSync) then begin
    syncIntf := CreateOmniCriticalSection;
    if CAS(nil, pointer(syncIntf), ocsSync) then
      pointer(syncIntf) := nil;
  end;
end;
```

4.1.3 Locked

`TOmniCS` is a great simplification of the critical section concept, but it still requires you to declare a separate locking entity. If this locking entity is only used to synchronize access to a specific instance (being that an object, record, interface or even a simple type) it is often better to declare a variable/field of type `Locked<T>` which combines any type with a critical section.

Using `Locked<T>`, the example from the `TOmniCS` section can be rewritten as follows.

```
uses
  GpLists,
  OtlSync;

procedure ProcessList(const intf: IGpIntegerList);
begin
  //...
end;

var
  lockedIntf: Locked<IGpIntegerList>;

procedure Test2;
begin
  lockedIntf := TGpIntegerList.CreateInterface;
  //...
```

```
17    lockedIntf.Acquire;
18    try
19      ProcessList(lockedIntf);
20    finally lockedIntf.Release; end;
21  end;
```

The interesting fact to notice is although the `lockedIntf` is declared as a variable of type `Locked<IGpIntegerList>`, it can be initialized and used as if it is of type `IGpIntegerList`. This is accomplished by providing `Implicit` operators for conversion from `Locked<T>` to `T` and back. Delphi compiler is (sadly) not smart enough to use this conversion operator in some cases so you would still sometimes have to use the provided `Value` property. For example, you'd have to do it to release wrapped object. (In the example above we have wrapped an interface and the compiler itself handled the destruction.)

```
1   procedure ProcessObjList(obj: TGpIntegerList);
2   begin
3     //...
4   end;
5
6   var
7     lockedObj: Locked<TGpIntegerList>;
8
9   procedure Test3;
10  begin
11    lockedObj := TGpIntegerList.Create;
12    try
13      //...
14      lockedObj.Acquire;
15      try
16        ProcessObjList(lockedObj);
17      finally lockedObj.Release; end;
18      //...
19    finally lockedObj.Value.Free; end;
20  end;
```

Besides the standard `Acquire`/`Release` methods, `Locked<T>` also implements methods used for *pessimistic locking*, which is described later in this chapter, and two almost identical methods called `Locked` which allow you to execute a code segment (a procedure, method or an anonymous method) while the critical section is acquired. (In other words, you can be assured that the code passed to the `Locked` method is always executed only once provided that all code in the program properly locks access to the shared variable.)

```
type
  Locked<T> = record
  public
    type TFactory = reference to function: T;
    type TProcT = reference to procedure(const value: T);
    constructor Create(const value: T; ownsObject: boolean = true);
    class operator Implicit(const value: Locked<T>): T; inline;
    class operator Implicit(const value: T): Locked<T>; inline;
    function  Initialize(factory: TFactory): T; overload;
    {$IFDEF OTL_ERTTI}
    function  Initialize: T; overload;
    {$ENDIF OTL_ERTTI}
    procedure Acquire; inline;
    procedure Locked(proc: TProc); overload; inline;
    procedure Locked(proc: TProcT); overload; inline;
    procedure Release; inline;
    procedure Free; inline;
    property Value: T read GetValue;
  end;

procedure Locked<T>.Locked(proc: TProc);
begin
  Acquire;
  try
    proc;
  finally Release; end;
end;

procedure Locked<T>.Locked(proc: TProcT);
begin
  Acquire;
  try
    proc(Value);
  finally Release; end;
end;
```

4.1.3.1 Why Not Use TMonitor?

There is an alternative built into Delphi since 2009 which provides functionality similar to the `Locked<T>` – `TMonitor`. In modern Delphis, **every** object can be locked by using `System.TMonitor.Enter` function and unlocked by using `System.TMonitor.Exit`. The example above could be rewritten to use the `TMonitor` without much work.

```
var
  obj: TGpIntegerList;

procedure Test4;
begin
  obj := TGpIntegerList.Create;
  try
    //...
    System.TMonitor.Enter(obj);
    try
      ProcessObjList(obj);
    finally System.TMonitor.Exit(obj); end;
    //...
  finally FreeAndNil(obj); end;
end;
```

A reasonable question to ask is, therefore, why implementing `Locked<T>`. Why is `TMonitor` not good enough? There are plenty of reasons for that.

- `TMonitor` was buggy since its inception[3] (although I believe that `Enter` and `Exit` may be stable enough for release code) and I don't like to use it.
- Using `TMonitor` doesn't convey your intentions. Just by looking at the variable/field declaration you wouldn't know that the entity is supposed to be used in a thread-safe manner. Using `Locked<T>`, however, explicitly declares your intent.
- `TMonitor.Enter/Exit` doesn't work with interfaces, records and primitive types. `Locked<T>` does.

4.2 TWaitFor

A common scenario in parallel programming is that the program has to wait for something to happen. The occurrence of that *something* is usually signalled with an event[4].

On Windows, this is usually accomplished by calling one of the functions from the WaitForMultipleObjects[5] family. While they are pretty powerful and quite simple to use, they also have a big limitation – one can only wait for up to 64 events at the same time.

Windows also offers a RegisterWaitForSingleObject[6] API call which can be used to circumvent this limitation. Its use is, however, quite complicated to use. To simplify programmer's life, OmniThreadLibrary introduces a `TWaitFor` class which allows the code to wait on any number of events.

[3] http://stackoverflow.com/questions/4856306/tthreadedqueue-not-capable-of-multiple-consumers, http://www.thedelphigeek.com/2011/05/tmonitor-bug.html

[4] https://en.wikipedia.org/wiki/Event_

[5] https://msdn.microsoft.com/en-us/library/windows/desktop/ms687025(v=vs.85).aspx

[6] https://msdn.microsoft.com/en-us/library/windows/desktop/ms685061(v=vs.85).aspx

```
type
  TWaitFor = class
  public type
    TWaitResult = (
      waAwaited,      // WAIT_OBJECT_0 .. WAIT_OBJECT_n
      waTimeout,      // WAIT_TIMEOUT
      waFailed,       // WAIT_FAILED
      waIOCompletion  // WAIT_IO_COMPLETION
    );
    THandleInfo = record
      Index: integer;
    end;
    THandles = array of THandleInfo;

    constructor Create; overload;
    constructor Create(const handles: array of THandle); overload;
    destructor  Destroy; override;
    function  MsgWaitAny(timeout_ms, wakeMask, flags: cardinal): TWaitResult;
    procedure SetHandles(const handles: array of THandle);
    function  WaitAll(timeout_ms: cardinal): TWaitResult;
    function  WaitAny(timeout_ms: cardinal; alertable: boolean = false): TWaitResult;
    property Signalled: THandles read FSignalledHandles;
  end;
```

To use `TWaitFor`, you have to create an instance of this class and pass it an array of handles either as a constructor parameter or by calling the `SetHandles` method. All handles must be created with the `CreateEvent` Windows function.

You can then wait for any (`WaitAny`) or all (`WaitAll`) events to become signalled. In both cases the `Signalled` array is filled with information about signalled (set) events. The `Signalled` property is an array of `THandleInfo` records, each of which (currently) only contains one field - an index (into the `handles` array) of the signalled event.

For example, if you want to wait on two events and then react to them, you should use the following approach:

```
var
  wf: TWaitFor;
  info: THandleInfo;

wf := TWaitFor.Create([handle1, handle2]);
try
  if wf.WaitAny(INFINITE) = waAwaited then begin
    for info in wf.Signalled do
      if info.Index = 0 then
        // handle1 is signalled - do something
      else if info.Index = 1 then
```

```
12          // handle2 is signalled - do something
13    end;
14  finally FreeAndNil(wf); end;
```

You don't have to recreate `TWaitFor` for each wait operation; it is perfectly ok to call `WaitXXX` functions repeatedly on the same object. It is also fine to change the array of handles between two `WaitXXX` calls by calling the `SetHandles` method.

The `WaitAny` method also comes in a variant which processes Windows messages, I/O completion routines and APC calls (`MsgWaitAny`). Its `wakeMask` and `flags` parameters are the same as the corresponding parameters to the MsgWaitForMultipleObjectsEx[7] API.

The use of the `TWaitFor` is shown in demo `59_TWaitFor`.

4.3 TOmniCounter

The `CreateCounter` (*OtlCommon* unit) function creates a counter with an atomic increment and decrement operations. Such counter can be used from multiple threads at the same time without any locking. Accessing the counter's value is also thread-safe.

The counter is returned as an `IOmniCounter` interface. It is implemented by the `TOmniCounter` class, which you can use in your code directly if you'd rather deal with objects than interfaces.

```
1   type
2     IOmniCounter = interface ['{3A73CCF3-EDC5-484F-8459-532B8C715E3C}']
3       function  Increment: integer;
4       function  Decrement: integer;
5       function  Take(count: integer): integer; overload;
6       function  Take(count: integer; var taken: integer): boolean; overload;
7       property Value: integer read GetValue write SetValue;
8     end;
9
10    TOmniCounter = record
11      procedure Initialize;
12      function  Increment: integer;
13      function  Decrement: integer;
14      function  Take(count: integer): integer; overload;
15      function  Take(count: integer; var taken: integer): boolean; overload;
16      property Value: integer read GetValue write SetValue;
17    end;
18
19    function  CreateCounter(initialValue: integer = 0): IOmniCounter;
```

The counter part of the `TOmniCounter` record is automatically initialized on the first use. If you want, you can call 'Initialize' in advance, although that is not required.

[7] https://msdn.microsoft.com/en-us/library/windows/desktop/ms684245(v=vs.85).aspx

`Take` is a special operation which tries to decrement the counter by `count` but stops at `0`. It returns the number that could be *taken* from the counter (basically, `Min(count, counter.Value)`). Its effect is the same as the following code (except that the real implementation of `Take` is thread-safe).

```
1  Result := Min(counter, count);
2  counter := counter - Result;
```

`Take` is used in demo *Parallel Data Production*.

4.4 TGp4AlignedInt and TGp8AlignedInt64

Two records – `TGp4AlignedInt` and `TGp8AlignedInt64` – are strictly speaking not part of the OmniThread-Library as they are implemented in the *GpStuff* unit, but they are used extensively through the library and as such are documented here.

Those two records hold 4-byte (32 bit) and 8-byte (64 bit) values, respectively. Those values are suitably aligned so they can be used in *interlocked* operations. Because of that, records can implement atomic `Increment`, `Decrement`, `Add`, and `Substract` operations.

Reading and writing values stored in the record (through the `Value` property or by using a supplied `Implicit` operator) is also atomic on the Win64 platform. On the Win32 platform, only the value of the `TGp4AligntInt` can be accessed atomically.

To read atomically from a `TGp8AlignedInt64` record on a Win32 platform, you can use the `CAS` (compare-and-swap) atomic operation.

```
1  type
2    TGp4AlignedInt = record
3    public
4      function  Add(value: integer): integer; inline;
5      function  Addr: PInteger; inline;
6      function  CAS(oldValue, newValue: integer): boolean;
7      function  Decrement: integer; overload; inline;
8      function  Decrement(value: integer): integer; overload; inline;
9      function  Increment: integer; overload; inline;
10     function  Increment(value: integer): integer; overload; inline;
11     function  Subtract(value: integer): integer; inline;
12     class operator Add(const ai: TGp4AlignedInt; i: integer): cardinal; inline;
13     class operator Equal(const ai: TGp4AlignedInt; i: integer): boolean; inline;
14     class operator GreaterThan(const ai: TGp4AlignedInt; i: integer): boolean; inline;
15     class operator GreaterThanOrEqual(const ai: TGp4AlignedInt;
16       i: integer): boolean; inline;
17     class operator Implicit(const ai: TGp4AlignedInt): integer; inline;
```

```
18      class operator Implicit(const ai: TGp4AlignedInt): cardinal; inline;
19      class operator Implicit(const ai: TGp4AlignedInt): PInteger; inline;
20      class operator LessThan(const ai: TGp4AlignedInt; i: integer): boolean; inline;
21      class operator LessThanOrEqual(const ai: TGp4AlignedInt; i: integer): boolean; inline;
22      class operator NotEqual(const ai: TGp4AlignedInt; i: integer): boolean; inline;
23      class operator Subtract(ai: TGp4AlignedInt; i: integer): cardinal; inline;
24      property Value read GetValue write SetValue: integer;
25    end;
26
27    TGp8AlignedInt64 = record
28    public
29      function  Add(value: int64): int64; inline;
30      function  Addr: PInt64; inline;
31      function  CAS(oldValue, newValue: int64): boolean;
32      function  Decrement: int64; overload; inline;
33      function  Decrement(value: int64): int64; overload; inline;
34      function  Increment: int64; overload; inline;
35      function  Increment(value: int64): int64; overload; inline;
36      function  Subtract(value: int64): int64; inline;
37      property Value: int64 read GetValue write SetValue;
38    end;
```

5. How-to

This part of the book contains practical examples of OmniThreadLibrary usage. Each of them starts with a question that introduces the problem and continues with the discussion of the solution.

Following topics are covered:

- *Background File Scanning*

Scanning folders and files in a background thread.

- *Async/Await*

How to write 'blocking' multithreaded code with a mechanism similar to .NET's `async/await`.

- *Web Download and Database Storage*

Multiple workers downloading data and storing it in a single database.

- *Parallel For with Synchronized Output*

Redirecting output from a parallel for loop into a structure that doesn't support multithreaded access.

- *Using taskIndex and Task Initializer in Parallel For*

Using `taskIndex` property and task initializer delegate to provide a per-task data storage in Parallel.For.

- *Background Worker and List Partitioning*

Writing server-like background processing.

- *Parallel Data Production*

Multiple workers generating data and writing it into a single file.

- *Building a Connection Pool*

Using OmniThreadLibrary to create a pool of database connections.

- *QuickSort and Parallel Max*

How to sort an array and how to process an array using multiple threads.

- *Parallel Search in a Tree*

Finding data in a tree.

- *Multiple Workers with Multiple Frames*

Graphical user interface containing multiple frames where each frame is working as a frontend for a background task.

- *OmniThreadLibrary and Databases*

Using databases from OmniThreadLibrary.

- *OmniThreadLibrary and COM/OLE*

Using COM/OLE from OmniThreadLibrary.

- *Using Message Queue with a TThread Worker*

Using OmniThreadLibrary's TOmniMessageQueue to communicate with a TThread worker.

5.1 Async/Await

> *.NET 4.5 introduced very useful asynchronous concepts - `async` and `await`. Is there a way of implementing these language constructs in Delphi?*

In short – they cannot be implemented, just emulated.

Before showing how to do that, let's return to the basics and see how `async`/`await` could be used if they existed in Delphi.

Let's assume you've inherited this pretty useless code.

```
procedure TForm125.Button1Click(Sender: TObject);
var
  button: TButton;
begin
  button := Sender as TButton;
  button.Caption := 'Working ...';
  button.Enabled := false;
  Sleep(5000);
  button.Enabled := true;
  button.Caption := 'Done!';
end;
```

Now, your boss says, you have to make it parallel so the user can start three copies of it. (You also have to add two new buttons to the form to start those instances but that's easy to do.)

There are many ways to solve this problem, some more and some less complicated; I'd like to point out a simplest possible solution. But first, let's take a detour into .NET waters...

.NET 4.5 introduced a heavy magical concept of 'async' and 'await'[1]. In short, it allows you to write a code like this:

```
procedure TForm125.Button1Click(Sender: TObject); async;
var
  button: TButton;
begin
  button := Sender as TButton;
  button.Caption := 'Working ...';
  button.Enabled := false;
  await CreateTask(
    procedure begin
      Sleep(5000);
    end);
  button.Enabled := true;
  button.Caption := 'Done!';
end;
```

[Please note that this is not a supported syntax; that's just an example of how the .NET syntax could look if Delphi would have supported it.]

The trick here is that `await` doesn't really wait. It relinquishes control back to the main loop which continues to process events etc. In other words – the rest of the program is running as usual. It may also call another asynchronous function and `await` on it. Only when an asynchronous function returns (any of them, if there are more than one running), the control is returned to the point of the appropriate `await` call and the code continues with the next line. [*Carlo Kok* wrote a nice article about how `await` works[2] on the *RemObjects Blog*.]

[1] http://blogs.msdn.com/b/pfxteam/archive/2012/04/12/10293335.aspx

[2] http://blogs.remobjects.com/blogs/ck/2012/08/08/p4690

Async/await needs extensive support of the compiler and there's absolutely no way to write an async/await clone in Delphi. But ... there's a simple trick which allows us to write the code in almost the same way. It uses OmniThreadLibrary's Async construct and the magic of anonymous methods.

```
procedure TForm125.Button1Click(Sender: TObject);
var
  button: TButton;
begin
  button := Sender as TButton;
  button.Caption := 'Working ...';
  button.Enabled := false;
  Parallel.Async(
    procedure begin
      Sleep(5000);
    end,
    Parallel.TaskConfig.OnTerminate(
      procedure begin
        button.Enabled := true;
        button.Caption := 'Done!';
      end));
end;
```

Async executes its parameter (the delegate containing the `Sleep` call) in a background thread. When this background task is completed, it executes the second parameter (the `OnTerminate` delegate) in the main thread. While the background task is working, the main thread spins in its own message loop and runs the user interface – just as it would in the .NET case.

With some syntactical sugar, you can fake a very convincing .NET-like behaviour.

> This form of `Async/await` is implemented in the *OtlParallel* unit (Async/Await).

```
type
  IAwait = interface
    procedure Await(proc: TProc);
  end;

  TAwait = class(TInterfacedObject, IAwait)
  strict private
    FAsync: TProc;
  public
    constructor Create(async: TProc);
    procedure Await(proc: TProc);
```

```
12    end;
13
14  function Async(proc: TProc): IAwait;
15  begin
16    Result := TAwait.Create(proc);
17  end;
18
19  { TAwait }
20
21  constructor TAwait.Create(async: TProc);
22  begin
23    inherited Create;
24    FAsync := async;
25  end;
26
27  procedure TAwait.Await(proc: TProc);
28  begin
29    Parallel.Async(FAsync, Parallel.TaskConfig.OnTerminated(
30    procedure begin
31      proc;
32    end));
33  end;
34
35  { TForm125 }
36
37  procedure TForm125.Button1Click(Sender: TObject);
38  var
39    button: TButton;
40  begin
41    button := Sender as TButton;
42    button.Caption := 'Working ...';
43    button.Enabled := false;
44    Async(
45      procedure begin
46        Sleep(5000);
47      end).
48    Await(
49      procedure begin
50        button.Enabled := true;
51        button.Caption := 'Done!';
52      end);
53  end;
```

To test, put three buttons on a form and assign the `Button1Click` handler to all three. Click and enjoy.

5.2 QuickSort and Parallel Max

I would like to sort a big array of data, but my comparison function is quite convoluted and sorting takes a long time. Can I use OmniThreadLibrary to speed up sorting?

On a similar topic – sometimes I'd also like to find a maximum data element in this big array, without doing the sorting. How would I approach this problem?

The answer to both parts of the problem is the same – use the Fork/Join abstraction.

5.2.1 QuickSort

The first part of this how-to implements a well-known quicksort[3] algorithm in a parallel way (see demo application `44_Fork-Join QuickSort` for the full code).

Let's start with a non-optimized single threaded sorter. This simple implementation is very easy to convert to the multithreaded form.

```
1  procedure TSequentialSorter.QuickSort(left, right: integer);
2  var
3    pivotIndex: integer;
4  begin
5    if right > left then begin
6      if (right - left) <= CSortThreshold then
7        InsertionSort(left, right)
8      else begin
9        pivotIndex := Partition(left, right, (left + right) div 2);
10       QuickSort(left, pivotIndex - 1);
11       QuickSort(pivotIndex + 1, right);
12     end;
13   end;
14 end;
```

As you can see, the code switches to an insertion sort when the dimension of the array drops below some threshold. This is not really important for the single threaded version (it only brings a small speedup) but it will help immensely with the multithreaded version.

Converting this quicksort to a multithreaded version is quite simple.

Firstly, we have to create a fork/join computation pool. In this example, it is stored in a global field.

[3] http://en.wikipedia.org/wiki/Quicksort

```
FForkJoin := Parallel.ForkJoin;
```

Secondly, we have to adapt the `QuickSort` method.

```
procedure TParallelSorter.QuickSort(left, right: integer);
var
  pivotIndex: integer;
  sortLeft  : IOmniCompute;
  sortRight : IOmniCompute;
begin
  if right > left then begin
    if (right - left) <= CSortThreshold then
      InsertionSort(left, right)
    else begin
      pivotIndex := Partition(left, right, (left + right) div 2);
      sortLeft := FForkJoin.Compute(
        procedure
        begin
          QuickSort(left, pivotIndex - 1);
        end);
      sortRight := FForkJoin.Compute(
        procedure
        begin
          QuickSort(pivotIndex + 1, right);
        end);
      sortLeft.Await;
      sortRight.Await;
    end;
  end;
end;
```

The code looks much longer but changes are really simple. Each recursive call to `QuickSort` is replaced with the call to `Compute` ...

```
sortLeft := FForkJoin.Compute(
  procedure
  begin
    QuickSort(left, pivotIndex - 1);
  end);
```

... and the code `Awaits` on both subtasks.

Instead of calling `QuickSort` directly, parallel version creates `IOmniCompute` interface by calling `FForkJoin.Compute`. This creates a subtask wrapping the anonymous function which was passed to the `Compute` and puts this subtask into the fork/join computation pool.

The subtask is later read from this pool by one of the fork/join workers and is processed in the background thread.

Calling `Await` checks if the subtask has finished its work. In that case, `Await` simply returns and the code can proceed. Otherwise (subtask is still working), `Await` tries to get one subtask from the computation pool, executes it, and then repeats from the beginning (by checking if the subtask has finished its work). This way, all threads are always busy either with executing their own code or a subtask from the computation pool.

Because two `IOmniCompute` interfaces are stored on the stack in each `QuickSort` call, this code uses more stack space than the single threaded version. That is the main reason why the parallel execution is stopped at some level and simple sequential version is used to sort remaining fields.

5.2.2 Parallel Max

The second part of this how-to finds a maximum element of an array in a parallel way (see demo application `45_Fork-Join max` for the full code).

The parallel solution is similar to the quicksort example above with few important differences related to the fact that the code must return a value (the quicksort code merely sorted the array returning nothing).

This directly affects the interface usage – instead of working with `IOmniForkJoin` and `IOmniCompute` the code uses `IOmniForkJoin<T>` and `IOmniCompute<T>`. As our example array contains integers, the parallel code creates `IOmniForkJoin<integer>` and passes it to the ParallelMax function.

```
max := ParallelMax(Parallel.ForkJoin<integer>, Low(FData), High(FData));
```

In this example fork/join computation pool is passed as a parameter. This approach is more flexible but is also slightly slower and – more importantly – uses more stack space.

```
function ParallelMax(
  const forkJoin: IOmniForkJoin<integer>;
  left, right: integer): integer;

var
  computeLeft : IOmniCompute<integer>;
  computeRight: IOmniCompute<integer>;
  mid         : integer;

  function Compute(left, right: integer): IOmniCompute<integer>;
  begin
    Result := forkJoin.Compute(
      function: integer
      begin
        Result := ParallelMax(forkJoin, left, right);
      end
    );
  end;
```

```
19
20  begin
21    if (right - left) < CSeqThreshold then
22      Result := SequentialMax(left, right)
23    else begin
24      mid := (left + right) div 2;
25      computeLeft := Compute(left, mid);
26      computeRight := Compute(mid + 1, right);
27      Result := Max(computeLeft.Value, computeRight.Value);
28    end;
29  end;
```

When the array subrange is small enough, `ParallelMax` calls the sequential (single threaded) version – just as the parallel QuickSort did, and because of the same reason – not to run out of stack space.

With a big subrange, the code creates two `IOmniCompute<integer>` subtasks each wrapping a function returning an integer. This function in turn calls back `ParallelMax` (but with a smaller range). To get the result of the anonymous function wrapped by the `Compute`, the code calls the `Value` function. Just as with the `Await`, `Value` either returns a result (if it was already computed) or executes another fork/join subtasks from the computation pool.

While creating fork/join programs, keep in mind this anti-pattern. The following code fragment is wrong!

```
1  Result := Max(Compute(left, mid).Value,
2   Compute(mid + 1, right).Value);
```

You must always create all subtasks before calling `Await` or `Value`! Otherwise, your code will not execute in parallel at all – it will all be processed by a single thread!

6. Appendix

6.1 Demo Applications

OmniThreadLibrary distribution includes plenty of demo applications that will help you get started. They are stored in the `tests` subfolder. This chapter lists all tests.

- `0_Beep` The simplest possible OmniThreadLibrary threading code.
- `1_HelloWorld` Threaded "Hello, World" with TOmniEventMonitor component created in runtime.
- `2_TwoWayHello` Hello, World with bidirectional communication; TOmniEventMonitor created in runtime.
- `3_HelloWorld_with_package` Threaded "Hello, World" with TOmniEventMonitor component on the form.
- `4_TwoWayHello_with_package` Hello, World with bidirectional communication; TOmniEventMonitor component on the form.
- `5_TwoWayHello_without_loop` Hello, World with bidirectional communication, the OTL way.
- `6_TwoWayHello_with_object_worker` Obsolete, almost totally equal to the demo `5_TwoWayHello_without_loop`.
- `7_InitTest` Demonstrates .WaitForInit, .ExitCode, .ExitMessage, and .SetPriority.
- `8_RegisterComm` Demonstrates creation of additional communication channels.
- `9_Communications` Simple communication subsystem tester.
- `10_Containers` Full-blown communication subsystem tester. Used to verify correctness of the lock-free code.
- `11_ThreadPool` Thread pool demo.
- `12_Lock` Demonstrates .WithLock.
- `13_Exceptions` Demonstrates exception catching.
- `14_TerminateWhen` Demonstrates .TerminateWhen and .WithCounter.
- `15_TaskGroup` Task group demo.
- `16_ChainTo` Demonstrates .ChainTo.
- `17_MsgWait` Demonstrates .MsgWait and Windows message processing inside tasks.
- `18_StringMsgDispatch` Calling task methods by name and address.
- `19_StringMsgBenchmark` Benchmarks various ways of task method invocation.
- `20_QuickSort` Parallel QuickSort demo.
- `21_Anonymous_methods` Demonstrates the use of anonymous methods as task workers in Delphi 2009.
- `22_Termination` Tests for .Terminate and .Terminated.
- `23_BackgroundFileSearch` Demonstrates file scanning in a background thread.
- `24_ConnectionPool` Demonstrates how to create a connection pool with OmniThreadLibrary.
- `25_WaitableComm` Demo for ReceiveWait and SendWait.
- `26_MultiEventMonitor` How to run multiple event monitors in parallel.

- `27_RecursiveTree` Parallel tree processing.
- `28_Hooks` Demo for the new hook system.
- `29_ImplicitEventMonitor` Demo for OnMessage and OnTerminated, named method approach.
- `30_AnonymousEventMonitor` Demo for OnMessage and OnTerminated, anonymous method approach.
- `31_WaitableObjects` Demo for the RegisterWaitObject/UnregisterWaitObject API.
- `32_Queue` Stress test for TOmniBaseQueue and TOmniQueue.
- `33_BlockingCollection` Stress test for the TOmniBlockingCollection, also demoes the use of Environment to set process affinity.
- `34_TreeScan` Parallel tree scan using TOmniBlockingCollection.
- `35_ParallelFor` Parallel tree scan using Parallel.ForEach (Delphi 2009 and newer).
- `37_ParallelJoin` ParallelJoin: Parallel.Join demo.
- `38_OrderedFor` Ordered parallel for loops.
- `39_Future` Futures.
- `40_Mandelbrot` Very simple parallel graphics demo.
- `41_Pipeline` Multistage parallel processes
- `42_MessageQueue` Stress test for TOmniMessageQueue.
- `43_InvokeAnonymous` Demo for IOmniTask.Invoke.
- `44_Fork-Join QuickSort` QuickSort implemented using Parallel.ForkJoin.
- `45_Fork-Join max` Max(array) implemented using Parallel.ForkJoin.
- `46_Async` Demo for Parallel.Async.
- `47_TaskConfig` Demo for task configuration with Parallel.TaskConfig.
- `48_OtlParallelExceptions` Exception handling in high-level OTL constructs.
- `49_FramedWorkers` Multiple frames each communication with own worker task.
- `50_OmniValueArray` Wrapping arrays, hashes and records in TOmniValue.
- `51_PipelineStressTest` Pipeline stress test by [Anton Alisov].
- `52_BackgroundWorker` Demo for the Parallel.BackgroundWorker abstraction.
- `53_AsyncAwait` Demo for the Async/Await abstraction.
- `54_LockManager` Lock manager (IOmniLockManager
- `55_ForEachProgress` Demonstrates progress bar updating from a ForEach loop.
- `56_RunInvoke` Simplified 'run & invoke' low-level API.
- `57_For` Simple and fast parallel for.
- `58_ForVsForEach` Speed comparison between Parallel.ForEach, Parallel.For, and TParallel.For (XE7+).
- `59_TWaitFor` Demo for the TWaitFor class.
- `60_Map` Demonstrates the Parallel.Map abstraction.
- `61_CollectionToArray` Demonstrates the TOmniBlockingCollection.ToArray method.

6.2 Examples

OmniThreadLibrary distribution includes some complex examples, stored in the `examples` subfolder. This chapter lists all examples. Many are also explained in the How-to chapter.

- `checkVat`

  OmniThreadLibrary and COM/OLE

  Using COM/OLE from OmniThreadLibrary.
- `forEach output`

  Parallel For with Synchronized Output

  Redirecting output from a parallel for loop into a structure that doesn't support multithreaded access.
- `report generator`

  Simulation of a report generator, which uses multiple background workers to generate reports; one worker per client.
- `stringlist parser`

  Background Worker and List Partitioning

  Writing server-like background processing.
- `TThread communication`

  Using Message Queue with a TThread Worker

  Using `TOmniMessageQueue` to communicate with a `TThread`-based worker.
- `twofish`

  OmniThreadLibrary and Databases

  Using databases from OmniThreadLibrary.